

Débuter en Perl

-
François Dagorn

-
Olivier Salaun

-
20 mars 2002

Résumé

Ce document est une présentation tutoriale du langage Perl. Il ne couvre pas tous les aspects du langage, sa lecture ne dispense pas de consulter les ouvrages de références cités dans la bibliographie.

Une version électronique est disponible à l'adresse suivante :
<http://perso.univ-rennes1.fr/Francois.Dagorn/perl>

Table des matières

1	Introduction	1
1.1	Généralités	1
1.2	Un aperçu de la syntaxe	2
2	Les types de données	3
2.1	Les scalaires	3
2.1.1	Les nombres	3
2.1.2	Les chaînes de caractères	4
2.1.3	Les variables scalaires	5
2.1.4	Interprétation des variables dans une chaîne de caractères	7
2.1.5	Des fonctions pour manipuler les scalaires	7
2.2	Les tableaux	8
2.2.1	Les éléments d'un tableau	8
2.2.2	Les variables tableaux	9
2.2.3	Des fonctions pour manipuler les variables tableaux	10
2.3	Les tableaux associatifs (hashes)	12

2.3.1	Les variables tableaux associatifs	12
2.3.2	Des fonctions adaptées aux tableaux associatifs	13
3	Les structures de contrôle	15
3.1	L'instruction if	16
3.2	L'instruction unless	17
3.3	Les instructions while et until	17
3.4	L'instruction for	17
3.5	L'instruction foreach	18
3.6	Exécuter si l'expression précédente est vraie ou fausse	19
4	Entrée standard et sortie standard	21
4.1	Lecture sur l'entrée standard	21
4.2	Écriture sur la sortie standard	22
5	La variable \$_	23
6	Les expressions régulières	25
6.1	L'opérateur de recherche d'occurrences d'expressions régulières	25
6.1.1	Recherche sur une variable quelconque	26
6.2	Construction des expressions régulières	26
6.2.1	Les sélecteurs de caractères	26
6.2.2	Les multiplicateurs de sélecteurs	27
6.2.3	Mise en mémoire d'une sélection partielle	28

6.2.4	La sélection alternative	29
6.2.5	Balisage des frontières de sélection	29
6.2.6	sélection d'un nombre exact d'occurrences	30
6.2.7	Priorité des opérateurs de construction d'expressions régulières	30
6.3	L'opérateur de substitution	31
6.4	Ignorer la casse	32
6.5	Travailler sur les champs d'une ligne	32
6.5.1	split	32
6.5.2	join	33
7	Quelques trucs utiles	35
7.1	Exécuter des commandes	35
7.2	La fonction die	36
7.3	Évaluation à la volée	36
7.4	Les arguments de la ligne de commande	37
8	La portée des variables	39
8.1	Déclarer des variables locales avec my	39
8.2	Une autre manière de déclarer des variables locales	40
8.3	use strict	41
9	Les fonctions	43
9.1	Définition d'une fonction	43
9.2	Appel d'une fonction	44

9.3	Prototyper les fonctions	45
9.4	Passer des arguments par références	47
9.4.1	Les références de variables	47
9.4.2	Un exemple	49
10	Accès au contenu des fichiers	51
10.1	Ouverture	51
10.2	Lecture	52
10.3	Écriture	53
10.4	Fermeture	53
11	Manipulations du système de gestion de fichiers	55
11.1	Accès aux répertoires	55
11.1.1	Utiliser la convention *	55
11.1.2	Utiliser une interface d'accès	56
11.2	Manipulation des fichiers et répertoires	56
11.2.1	Changer de répertoire de travail	57
11.2.2	Créer un répertoire	57
11.2.3	Supprimer un répertoire	57
11.2.4	Supprimer un fichier	57
11.2.5	Renommer un fichier	58
11.2.6	Modifier les droits d'accès	58
11.3	Fonctions utiles	58

12 Les structures de données complexes	61
12.1 Les listes de listes	61
12.2 Les hashes de hashes	63
12.3 Autres structures de données	64
13 Le débogueur de Perl	67
14 Écriture et utilisation de modules	71
14.1 Inclure du code	71
14.2 Les packages	73
14.3 Remarques	75
15 Écriture orientée objet	77
15.1 Un exemple limité à l'usage classique des packages	77
15.2 Référencer une classe	79
15.3 Hériter d'une classe	80
16 L'écriture de scripts CGI	83
16.1 Introduction	83
16.2 Un exemple	84
16.2.1 Utiliser cgi-lib.pl	85
16.2.2 Utiliser CGI.pm	87
16.3 Envoyer un fichier (<i>upload</i>)	88
16.4 Utiliser FastCGI	90

16.4.1	Introduction	90
16.4.2	Un exemple	90
17	La programmation d'applications réseaux	93
17.1	Introduction	93
17.2	Un exemple d'application client/serveur	93
	Bibliographie	97

Chapitre 1

Introduction

1.1 Généralités

Perl est un langage de programmation adapté au traitement des fichiers textes ainsi qu'aux tâches d'administration des systèmes et des applications. Disponible dans le domaine public pour les environnements Unix, Linux, Windows ... il connaît aujourd'hui un développement très important.

Créé en 1986 par Larry Wall, Perl a depuis connu de nombreuses versions, aujourd'hui on utilise Perl5, la version 5.6¹ est maintenant définitive.

Perl est un langage interprété et les programmes Perl se diffusent en format source. Le terme interprété est toutefois ambigu car un programme Perl s'exécute en deux phases : la pré-compilation du programme qui produit du pseudo-code, puis l'interprétation du pseudo-code (l'exécution proprement dite). Le pseudo-code de Perl contient plus de 300 méta-instructions, à l'issue de la pré-compilation on obtient une suite de pseudo-codes autonomes qui disposent de pointeurs vers leurs opérands et la méta-instruction suivante.

Perl est très référencé sur le Web, parmi les principaux sites citons :

- <http://www.perl.org> (tout sur Perl)
- <http://www.perl.com> (tout sur Perl également)
- <http://www.cpan.org> (distribution de Perl et des modules)

1. <http://www.perl.com/pub/language/info/software.html#stable>

1.2 Un aperçu de la syntaxe

Perl est sensible à la casse (les majuscules et minuscules sont différenciées), c'est à dire que **print** est une fonction du langage, alors que **Print** ne l'est pas, mais peut bien sûr être un identificateur (de variables, de fonction, de fichier) à la charge du programmeur.

Il n'y a pas de formatage particulier dans l'écriture d'une instruction Perl, le séparateur peut être un espace, une tabulation ou même un retour à la ligne. Les instructions sont séparées entre elles par le caractère ; (le point virgule) et un commentaire est introduit en plaçant le caractère # à n'importe quel endroit d'une ligne (mais il ne s'étend que sur une ligne).

Un programme Perl (souvent appelé **script**) débute en général par une ligne qui indique l'endroit où se trouve le compilateur/interpréteur², c'est ainsi que l'on peut sur des systèmes Unix démarrer une application Perl en citant simplement son nom. Il n'est pas inopportun de procéder de même sur des systèmes Windows³, du moins dans un environnement WWW/CGI (cf. section 16) car des serveurs http (tel APACHE) savent tenir compte de cette première ligne et lancer Perl en fonction de ce qu'elle indique.

2. `#!/usr/local/bin/perl` par exemple ...

3. Sur lesquels on lance une application Perl en la citant derrière le chemin d'accès à l'interpréteur ou en effectuant une association entre un suffixe de fichier (**.pl** par exemple) et l'interpréteur Perl.

Chapitre 2

Les types de données

En Perl il n'existe que 3 types de variables: les scalaires, les tableaux et les tableaux associatifs (*hash*).

2.1 Les scalaires

Une variable de type scalaire est un nombre ou une chaîne de caractères. Il n'existe pas de type particulier pour les valeurs entières, les nombres réels, les caractères ... Lorsque dans une expression figurent des variables scalaires correspondants à des valeurs de genres différents, Perl effectue des conversions totalement transparentes (et dangereuses).

2.1.1 Les nombres

Les valeurs littérales numériques qui peuvent être affectées à une variable scalaire s'expriment classiquement :

<code>12</code>	(une valeur entière positive)
<code>+10</code>	(une autre valeur entière positive)
<code>-34</code>	(une valeur entière négative)
<code>+3.14</code>	(un réel positif)
<code>5e15</code>	(5 fois 10 puissance 15)
<code>033</code>	(33 en code octal soit 27 en décimal)
<code>x1F</code>	(1F en hexa soit 31 en décimal)

Les opérateurs pour les nombres

Les opérateurs arithmétiques sont les suivants :

<code>+</code>	(addition)	<code>-</code>	(soustraction)
<code>*</code>	(multiplication)	<code>**</code>	(puissance)
<code>/</code>	(division)	<code>%</code>	(modulo)

Les opérateurs de comparaison de nombres sont les suivants :

<code><</code>	(inférieur)	<code>></code>	(supérieur)
<code><=</code>	(inférieur ou égal)	<code>>=</code>	(supérieur ou égal)
<code>==</code>	(égalité)	<code>!=</code>	(différence)

2.1.2 Les chaînes de caractères

Une chaîne de caractères Perl comprend de 0 à n caractères ASCII dont le code est compris entre 0 et 255. Il n'y a donc pas de restriction sur le caractère NUL ou sur les autres caractères non imprimables. Il y a deux manières de spécifier la valeur littérale d'une chaîne :

- utiliser une **simple quote** comme délimiteur, le contenu est pris tel que, il n'y a pas d'interprétation de caractères particuliers. Dans ce cas pour introduire une simple quote dans la chaîne, il convient de la précéder d'un antislash (`\`) , pour introduire un antislash, il faut le doubler.

<code>'Bonjour'</code>	une chaîne
<code>'Bonjour \n'</code>	le <code>\n</code> n'a pas de sens ici
<code>'L\'école'</code>	le <code>\</code> devant <code>'</code> pour l'apostrophe
<code>'c:\\windows\\system'</code>	antislash doublés
<code>''</code>	chaîne vide

- utiliser une **double quote** comme délimiteur, certaines séquences de caractères seront interprétées. L'antislash (`\`) est utilisé ici comme caractère d'échappement et s'utilise comme suit :

"\n"	une nouvelle ligne
"\t"	une tabulation
"\033"	un caractère codé en octal (33 ici)
"\x1F"	un caractère codé en hexa (1F ici)
"\cA"	un caractère de Contrôle (Ctrl-A ici)
"\""	un \
"\""	une double quote

Les chaînes de caractères présentées entre double quotes permettent également d'interpréter le contenu de variables (cf. 2.1.4).

Les opérateurs pour les chaînes de caractères

L'opérateur de concaténation est le . (le point) :

"Bonjour"."Monsieur"	<==>	"BonjourMonsieur"
'Il'.' ' . "fait beau \n"	<==>	"Il fait beau \n"

Les opérateurs de comparaison de chaînes de caractères sont les suivants :

eq (égalité)	ne (différence)
lt (inférieur)	gt (supérieur)
le (inférieur ou égal)	ge (supérieur ou égal)

Les opérateurs **q** et **qq** permettent de remplacer respectivement les délimiteurs de chaîne **quote** et **double quote** par un délimiteur au choix du programmeur :

\$chaine='S'il vous plait';
peut aussi s'écrire
\$chaine=q@s'il vous plait@;
C'est le caractère @ qui est ici utilisé comme délimiteur

2.1.3 Les variables scalaires

Une variable scalaire est représentée par une suite de caractère précédée du symbole \$. Le premier caractère suivant le \$ doit être une lettre, ensuite on peut trouver un nombre

quelconque de chiffres et de lettres. Les majuscules et minuscules sont différenciées et le caractère _ (souligné) est autorisé :

<code>\$Une_Variable</code>	un nom de variable correct
<code>\$une_variable</code>	un autre nom de variable
<code>\$une-variable</code>	un nom de variable incorrect (- vs. _)

L'affectation des variables scalaires

L'opérateur d'affectation en Perl est le signe =

<code>\$I = 10;</code>	(affecte la constante 10 à la variable I)
<code>\$K = \$I + 1;</code>	(affectation de I + 1 à la variable K soit 11)
<code>\$chaine = 'Bonjour' . 'à tous';</code>	(affectation d'une chaîne)

Les programmeurs SH ou CSH noteront que pour référencer une variable, il convient de précéder son nom par le symbole \$ de part et d'autre de l'opérateur d'affectation (le signe =).

Il existe d'autres opérateurs d'affectation, ce sont des raccourcis d'écriture familiers aux programmeurs C :

– affectation et opération combinée

<code>\$var += 2;</code>	<code><==> \$var = \$var + 2;</code>
<code>\$var *= 3;</code>	<code><==> \$var = \$var * 3;</code>
<code>\$chaine .= 'xxx';</code>	<code><==> \$chaine = \$chaine . 'xxx';</code>

– Incrémentation et décrémentation automatique

<code>\$I = 10;</code>	
<code>\$I++;</code>	<code><==> \$I = \$I + 1; donc I = 11</code>
<code>\$K = ++\$I;</code>	<code><==> \$K = \$I = \$I + 1; donc K = I = 12</code>
<code>\$K = \$I++;</code>	<code><==> \$K = \$I; \$I = \$I + 1;</code> <code>donc K = 12 et I = 13</code>

L'incrément se fait avant l'affectation si l'opérateur précède le nom de la variable (`++$I`) ou après l'affectation si l'opérateur suit le nom de la variable (`$I++`).

2.1.4 Interprétation des variables dans une chaîne de caractères

Des variables scalaires peuvent être interprétées lorsqu'elles sont contenues dans des chaînes de caractères présentées entre double quotes. À cet effet, Perl recherche le caractère `$` et considère ce qui le suit comme un nom de variable potentiel. Si la variable existe, son contenu est intégré à la chaîne source, si elle n'existe pas, elle est remplacée par une chaîne vide. Il n'y a qu'un seul niveau d'interprétation, si une variable contient elle-même une chaîne faisant référence à une autre variable, il n'y aura aucun traitement. Pour empêcher l'interprétation comme un nom de variable des caractères qui suivent le caractère `$`, il convient de le précéder d'un antislash ou de placer le morceau de chaîne considéré entre simple quotes.

```
$chaine1 = "Au revoir";
$chaine2 = "$chaine1 les petits amis";
$chaine3 = "au plaisir ";
$chaine4 = "${chaine3}de vous revoir"; Les accolades sont
           utilisées ici comme en sh, csh pour indiquer
           explicitement la variable à interpréter.
```

2.1.5 Des fonctions pour manipuler les scalaires

En plus des opérateurs du langage, Perl propose des fonctions (cf. section 9) adaptées au traitement des variables scalaires, parmi celles-ci citons :

- **chop()** enlève le dernier caractère d'une chaîne;
- **chomp()** enlève le dernier caractère d'une chaîne si celui-ci est un délimiteur de fin de ligne;
- **chr()** convertit une valeur entière en le caractère correspondant du code ASCII;
- **hex()** renvoie la valeur décimale correspondant à une chaîne de caractères hexadécimaux;
- **oct()** renvoie la valeur décimale correspondant à une chaîne de caractère octaux;
- **index()** renvoie la position de la première occurrence d'une sous chaîne dans une chaîne;
- **lc()** convertit une chaîne de caractères en caractères minuscules;
- **length()** indique la longueur en caractère d'une variable scalaire;
- **rindex()** renvoie la position de la dernière occurrence d'une sous chaîne dans une chaîne;
- **substr()** permet d'extraire d'une chaîne une sous chaîne définie en position et longueur.

```

$Esc = chr(27);      # convertit l'entier 27 en ASCII
$B10 = hex("1F3C"); # $B10 vaut 7996
$B10 = oct("1073"); # $B10 vaut 571
$Min = lc("ABCDE"); # $Min vaut "abcde"
$Lg  = length($Min); # $Lg vaut 5 (longueur en caractères)
$Lg  = length($B10); # $Lg vaut 3 d'un scalaire)

```

2.2 Les tableaux

Une variable de type tableau est une liste de données scalaires (nombres et/ou chaînes). Chaque élément du tableau est une variable scalaire accessible comme telle.

2.2.1 Les éléments d'un tableau

Les valeurs littérales qui peuvent être affectées à une variable de type tableau s'expriment comme une liste de valeurs séparées par des virgules et encadrées par des parenthèses.

```

()                <==> tableau vide
(8,4,9,4,5)      <==> 5 éléments (des nombres)
('toto',"$var",3) <==> 3 éléments (la chaîne 'toto',
la valeur courante de la variable $var et le nombre 3)

```

Un opérateur de construction liste (..) peut être utilisé pour spécifier un élément de tableau.

```

(1..10)          <==> 10 éléments (les chiffres de 1 à 10)
(1..5,10,20..30) <==> 16 éléments (les chiffres de 1 à 5,
                        puis le chiffre 10 et les 10 chiffres
                        de 20 à 30)
($debut..$fin)  <==> (($fin+1) - $debut) éléments ayant
                        pour valeurs de $debut à $fin par incrément de 1

```

2.2.2 Les variables tableaux

Le nommage d'une variable tableau suit les mêmes conventions que celui d'une variable scalaire, hormis le premier caractère qui est ici @ (un arobas). C'est le premier caractère qui détermine le type de la variable utilisée, si @**tab** est une variable de type tableau, elle n'a rien à voir avec \$**tab** qui est une variable de type scalaire. Une variable de type tableau qui n'a jamais été affectée a pour valeur une liste vide ().

L'affectation des variables tableaux

Le signe égal est l'opérateur d'affectation des variables tableaux.

```
$I = 5;
@tab1 = ();           <==> @tab1 est vide
@tab2 = (5,3,15);    <==> @tab2 contient 3 éléments de
                        valeurs 5, 3 et 15
@tab1 = @tab2;       <==> @tab1 est une copie de @tab2
@tab2 = $I;          <==> @tab2 reçoit un scalaire et
                        ne contient plus qu'un seul élément (de valeur 5)
($a,$b,$c) = @tab1; <==> $a = 5; $b = 3; $c = 15;
```

accès aux éléments d'une variable tableau

Les éléments constitutifs d'une variable de type tableau sont des scalaires, la référence à un élément commence donc par le symbole \$, un indice exprimé entre crochets ([]) indique l'occurrence ciblée. Le premier élément d'un tableau est accessible par l'indice 0.

```
$I = 5;
@tableau = ('x',2,3,4,5,'x');
$tableau[0] = 1;      <==> accès au 1er élément du tableau
$tableau[$I] = 6;    <==> accès au Ième élément
```

L'indice du dernier élément d'une variable tableau est accessible par la variable scalaire \$#**nom-du-tableau**. Si on affecte un élément au delà de cette limite, les *trous* sont

bouchés par des éléments de valeur **undef**¹.

```
@tab = (10,20,30);
$stab[$#tab+1] = 40; <==> $stab[3] = 40;
                        $#tab vaut maintenant 3
$stab[10] = 1000;      <==> $#tab vaut maintenant 10, les
                        éléments intermédiaires $stab[4]..$stab[9]
                        prennent la valeur undef
```

L'affectation d'une variable tableau dans une variable scalaire rend le nombre d'éléments de la variable tableau :

```
@tab = (1,2,3,4);
$scal = @tab;
# $scal vaut 4
```

2.2.3 Des fonctions pour manipuler les variables tableaux

Diverses fonctions permettent d'effectuer des décalages, des tris ... sur des listes (des variables tableaux).

Les fonctions **push()** et **pop()**

Les fonctions **push()** et **pop()** permettent de travailler sur l'extrémité droite d'une liste (ajout ou suppression d'éléments) :

```
@liste=('a','b');
push(@liste,('c','d','e')); <==> @liste=('a','b','c','d','e')
pop(@liste);                <==> @liste=('a','b','c','d')
```

Les fonctions **shift()** et **unshift()**

1. **undef** vaut 0 pour un nombre ou vide pour une chaîne. On peut indifféremment utiliser les opérateurs de comparaison **==** ou **eq** pour comparer un scalaire à la valeur **undef**

Les fonctions **shift()** et **unshift()** permettent de travailler sur l'extrémité gauche d'une liste (ajout ou suppression d'éléments) :

```
@liste=('c','d','e');
unshift(@liste,('a','b')); <==> @liste=('a','b','c','d','e')
shift(@liste);                <==> @liste=('b','c','d','e')
```

La fonction sort()

La fonction **sort()** permet de trier une liste dans l'ordre ASCII de ses éléments (les nombres sont convertis en chaînes avant le tri) :

```
@liste=(100,1,2,58,225);
@tri=sort(@liste); <==> @tri=('1','100','2','225','58')
@liste=('rouge','vert','bleu');
@tri=sort(@liste); <==> @tri=('bleu','rouge','vert')
```

La fonction reverse()

La fonction **reverse()** permet d'inverser les éléments d'une liste :

```
@liste=(1,2,3,4,5);
@rev=reverse(@liste); <==> @rev=(5,4,3,2,1)
```

La fonction chop()

La fonction **chop()** permet de supprimer le dernier caractère de tous les éléments d'une liste. Elle fonctionne également sur une variable scalaire et est surtout utilisée pour supprimer les caractères de fin de lignes contenus dans des lignes (résultats de saisies ...) :

```
@liste=("tutu\n","toto\n");
chop(@liste); <==> @liste=("tutu","toto")
```

La fonction chomp()

La fonction **chomp()** permet de supprimer le dernier caractère (seulement s'il s'agit d'un délimiteur de fin de ligne) de tous les éléments d'une liste. Il est donc préférable d'utiliser **chomp()** pour supprimer les éventuels délimiteurs de fin de lignes résultant d'une saisie, d'une lecture de fichier, ...

La fonction qw()

L'affectation de constantes littérales dans un tableau est fréquente et d'une écriture un peu fastidieuse, Perl propose la fonction **qw()** qui travaille sur une liste de mots :

```
@liste=("tutu","toto",'machin'); # est équivalent à
@liste = qw(tutu toto machin);   # ou même à
@liste = qw(tutu
            toto
            machin);
```

2.3 Les tableaux associatifs (hashes)

Un tableau associatif (ou hash) est un tableau dont les éléments (des scalaires) sont accédés au moyen d'un index qui est une valeur scalaire quelconque. Les éléments d'un hash sont des couples (clef, valeur).

2.3.1 Les variables tableaux associatifs

Le nommage d'un tableau associatif suit les mêmes conventions que celui d'une variable scalaire, hormis le premier caractère qui est ici un % (le caractère pour-cent).

```
%trad = ('january','janvier','february','avril');

# On affecte le hash %trad par une liste qui contient un
# nombre pair d'éléments correspondant à des couples
# clef, valeur.

$trad{'february'}='fevrier';
# l'élément du hash %trad dont la clef est 'february' reçoit
# une nouvelle valeur.
```

Une autre forme peut être utilisée pour spécifier les valeurs d'un tableau associatif, elle met en évidence la correspondance clef/valeur :

```
%trad = ( 'january' => 'janvier',
          'february' => 'fevrier',
          'march'    => 'mars' );
```

2.3.2 Des fonctions adaptées aux tableaux associatifs

keys

La fonction `keys(%var_hashee)` rend une liste des clefs d'un hash.

```
%hash = ('un',1,'deux',2,'trois',3,'quatre',4);
@liste_clef = keys(%hash);
# <==> @liste_clef = ('un','deux','trois','quatre');
```

values

La fonction `values(%var_hashee)` rend une liste des valeurs d'un hash.

```
%hash = ('un',1,'deux',2,'trois',3,'quatre',4);
@liste_val = values(%hash);
# <==> @liste_val = (1,2,3,4);
```

each

La fonction `each(%var_hashee)` rend un couple clef,valeur dans une liste à deux éléments. Chaque appel de `each` sur une même variable de type hash rend le couple suivant.

```
%hash = ('un',1,'deux',2,'trois',3,'quatre',4);
($clef,$valeur) = each(%hash);
# au 1er appel <==> $clef='un'; et $valeur=1;
```

each est la fonction adaptée pour parcourir entièrement un hash, elle rend une liste vide lorsque la fin du hash est atteinte.

delete

La fonction **delete** permet de supprimer un élément d'un hash en indiquant sa clef.

```
%hash = ('un',1,'deux',2,'trois',3,'quatre',4);  
delete($hash{'deux'});
```

Chapitre 3

Les structures de contrôle

Diverses structures de contrôle sont proposées en Perl, elles évaluent une expression et proposent un traitement selon les schémas algorithmiques classiques. L'expression à évaluer est convertie en chaîne, une chaîne non vide ou différente de "0" correspond à la valeur **vrai**. Les opérateurs de comparaison (==,>,gt,...) retournent 1 pour vrai et 0 pour faux ("0" lorsque converti en chaîne), ci-dessous quelques exemples d'expressions vraies ou fausses :

```
0           <==> faux
"0"        <==> faux, c'est 0 converti en chaîne
''         <==> faux, chaîne vide
'00'       <==> vrai, différent de 0 ou '0'
1          <==> vrai
"n'importe quoi" <==> vrai
undef      <==> undef rend une chaîne vide donc faux
```

Perl permet de structurer des instructions en utilisant des blocs délimités comme en C par les accolades ({}). La structuration en blocs a un impact sur la portée des variables (cf. section 8).

3.1 L'instruction if

La syntaxe générale de l'instruction **if** est la suivante :

```
if (expression_à_évaluer) {
    instruction(s) à exécuter si l'expression est vraie
} else {
    instructions à exécuter si l'expression est fausse
}
```

Si il n'y a rien de particulier à faire dans le cas où l'expression est fausse, on peut se passer du **else {...}**. Par contre même s'il n'y a qu'une seule instruction à exécuter, il convient de la placer dans un bloc **{}**.

```
if ($I == 0) { $I++; }          <==> le else est omis
if ((10 == 10) == 1) { ... } <==> vrai car == rend 1 si la
                                condition est réalisée
```

Le test de plusieurs cas peut être enchaîné en utilisant **elsif** :

```
if ($Couleur eq "Bleu") { ... }
elsif ($Couleur eq "Rouge") { ... }
    elsif ($Couleur eq "Vert") { ... }
        else { ... }

if ($Couleur eq "Noir") { ... }
else if ($Couleur eq "Jaune") <==> erreur de syntaxe,
    un else doit être suivi par une accolade {
```

3.2 L'instruction unless

L'instruction **unless** fonctionne comme un **nif** (non if: si l'expression est fausse alors faire).

```
unless ($I > 0) { ... }
else { ... }
```

Un **elsif** peut suivre un **unless**, le **elseunless** n'existe pas.

3.3 Les instructions while et until

Perl dispose d'une structure *tant que* (**while**) et d'une structure *jusqu'à ce que* (**until**).

```
$I = 0;
while ($I < 10) { $I++; } <==> tant que I est
                           inferieur à 10
$I = 0
until ($I == 10) { $I++; } <==> jusqu'à ce que
                              I soit égal à 10
while (1) { ... }           <==> sans fin
while ( ) { ... }          <==> sans fin, une expression
                           inexistante n'est pas vide !
until (0) { ... }         <==> sans fin
```

Pour forcer la sortie d'une boucle **while** ou **until** sans s'occuper de la condition d'arrêt, on utilise l'instruction **last** (de la même façon qu'un **break** en Langage C). Pour faire évaluer la condition sans avoir déroulé la totalité des instructions du bloc, on utilise l'instruction **next**.

3.4 L'instruction for

L'instruction **for** correspond à une boucle *pour* dans laquelle on fournit une valeur de départ, une expression à évaluer et une expression permettant une réinitialisation (incrément,

décrément, ...).

```
for ($I=0; $I < 10; $I++) { ... }
```

Dans l'exemple ci-dessus, on affecte 0 à I, on évalue ($I < 10$), si c'est vrai on exécute les instructions incluses dans le bloc et on passe à l'incrément de I, si c'est faux on passe à l'instruction suivante.

```
for (;;) { ... } <==> boucle sans fin, les expressions
                    ne sont pas obligatoires
```

Les instructions **next** et **last** fonctionnent de la même manière que pour **while** et **until** (cf. section 3.3).

3.5 L'instruction foreach

L'instruction **foreach** charge dans une variable les valeurs successives d'une liste et effectue le traitement spécifié dans le bloc d'instructions. Quand la fin de la liste est atteinte, on passe à l'instruction suivante :

```
foreach $Couleur ('jaune','vert','bleu') { ... }
    la variable couleur prend successivement les
    valeurs jaune, vert et bleu
foreach $var (@tableau1, @tableau2, $I, $K, 10)
    la variable $var prend successivement toutes les
    valeurs scalaires contenues dans $tableau1 ...
    jusqu'à la valeur 10
```

Ici aussi, **last** provoque une sortie forcée de la boucle et **next** un passage à la valeur suivante.

3.6 Exécuter si l'expression précédente est vraie ou fausse

L'opérateur **&&** permet d'évaluer une expression si celle qui la précède est vraie, l'opérateur **||** permet d'évaluer une expression si celle qui la précède est fausse :

```
$J = $I % 2 && printf ("I est impair \n");

    I modulo 2 vaut 0 ou 1, si 1 le nombre est impair
                                si 0 le nombre est pair

$J = $I % 2 || printf ("I est pair \n");
```

Les opérateurs **&&** et **||** peuvent être utilisés pour former des expressions conditionnelles évaluées par l'instruction **if** :

```
if (($a == $b) || ($a == $c)) { ... }
```

Une dernière construction est possible, elle admet 3 expressions séparées par les opérateurs **?** et **:**, la seconde est évaluée si la première est vraie, la troisième est évaluée si la première est fausse :

```
$J = $I % 2 ? printf ("impair \n") : printf ("Pair \n");
```


Chapitre 4

Entrée standard et sortie standard

Compte tenu du passé Unix de Perl, on lit sur le STDIN et on écrit sur le STDOUT. Par défaut il s'agit du clavier et de l'écran, mais Perl ne s'en occupe pas, c'est au système d'exploitation de lui fournir des descripteurs valides.

4.1 Lecture sur l'entrée standard

L'opérateur de lecture sur l'entrée standard est `<STDIN>`.

<pre>\$ligne = <STDIN>;</pre>	<pre><==> on affecte à la variable \$ligne la ligne suivante</pre>
<pre>@lignes = <STDIN>;</pre>	<pre><==> le tableau @lignes reçoit un élément par lignes lues</pre>

Pour lire et traiter ligne à ligne on procède en général comme suit :

<pre>while (\$ligne = <STDIN>) { ... }</pre>	<pre>Lorsque la dernière ligne est atteinte, <STDIN> retourne undef et la boucle while s'arrête</pre>
--	---

Perl autorise une autre forme de lecture du STDIN avec l'opérateur `<>` qui passe au programme tous les fichiers indiqués sur la ligne de commande (chaque argument de la ligne de commande est alors considéré comme un nom de fichier).

4.2 Écriture sur la sortie standard

L'instruction **print** écrit une liste de chaîne de caractères sur la sortie standard.

```
print (@liste1, $chaine, @liste2, 'toto');
```

Il est possible de contrôler la présentation des impressions sur la sortie standard en utilisant l'instruction **printf** qui fonctionne comme en C. Elle admet comme argument une liste dont le premier élément est une chaîne de caractères qui contient le masque d'impression des variables ainsi que des valeurs constantes à intercaler entre elles (les autres éléments de la liste sont les variables à imprimer).

```
printf ("%s %s %5d \n", $chaine1, $chaine2, $I);
%s est le masque d'impression de $chaine1
%s est le masque d'impression de $chaine2
%5d est le masque d'impression de $I
\n constante qui introduit une nouvelle ligne
```

Si un argument de **printf** est une variable tableau, il convient de spécifier autant de masques d'impression que de valeurs scalaires distinctes contenues dans le tableau. Les masques d'impression indiquent les conversions à réaliser et s'utilisent comme dans l'exemple suivant :

<code>%10s</code>	<code><==></code>	une chaîne sur sa longueur réelle complétée éventuellement par des espaces (10 maxi)
<code>%.5s</code>	<code><==></code>	les 5 premiers caractères d'une chaîne
<code>%3d</code>	<code><==></code>	un entier tel que ou sur 3 caractères s'il est inférieur à 99 (espace à gauche)
<code>%o</code>	<code><==></code>	un nombre en octal
<code>%x</code>	<code><==></code>	un nombre en hexadécimal
<code>%f</code>	<code><==></code>	un réel avec 6 décimales
<code>%.3f</code>	<code><==></code>	un réel avec 3 décimales
<code>%e</code>	<code><==></code>	un réel en représentation mantisse/exposant

Chapitre 5

La variable `$_`

Perl utilise une variable scalaire fourre-tout nommée `$_` (`$` souligné) comme résultat par défaut d'un certain nombre d'instructions. C'est notamment le cas avec la lecture sur l'entrée standard :

```
while (<STDIN>)          <==>  while ($_ = <STDIN>)
{                          <==>  {
    chop;                  <==>    chop($_);
}                          <==>  }
```

La variable `$_` est également implicitement utilisée par les instructions **foreach** et **print**¹, ainsi que par les instructions qui manipulent des expressions régulières (cf. section 6) ou celles qui testent l'existence de fichiers et répertoires (cf. section 11.3).

```
@liste=(1,2,3,4,5);
foreach (@liste)        <==>  foreach $_ (@liste)
{                          <==>  {
    print ;                <==>    print $_;
}                          <==>  }
```

1. Mais **printf** n'utilise pas la variable `$_` par défaut.

6.1.1 Recherche sur une variable quelconque

Lorsque la variable à osculer n'est pas \$_, on utilise l'opérateur =~ :

```
$var = "Il fait beau";
if ($var =~ /fait/)
{
    print ($&,$',,$',"\\n"); # <==> $& = 'fait', $\'='Il '
                            # <==> $' = ' beau'
}
```

6.2 Construction des expressions régulières

6.2.1 Les sélecteurs de caractères

Des sélecteurs de caractères sont utilisés pour construire les expressions régulières. Les plus simples recherchent l'existence d'un caractère, ils peuvent être répétés pour construire des expressions plus compliquées :

.	recherche n'importe quel caractère (sauf le changement de ligne)
[abc]	recherche un caractère parmi ceux situés entre les accolades (classe de caractères)
[^abc]	recherche un caractère qui ne soit pas un de ceux situés entre les accolades
[a-z]	recherche un caractère dont le code ASCII est situé entre le 1er et le 2ème caractère cité
ab.[a-z]	dans cet exemple, on recherche un a, suivi d'un b, suivi d'un caractère quelconque, suivi d'une lettre minuscule

Il existe des classes de caractères prédéfinies :

- (**\d**) correspond à un chiffre donc à **[0-9]** ;
- (**\w**) correspond à une lettre (plus le caractère souligné !) donc à **[a-zA-Z0-9_]** ;
- (**\s**) correspond aux caractères de séparation usuels (espace, tabulation, retour charriot, nouvelle ligne, saut de page).

Ces 3 classes de caractères ont une construction négative **\D**, **\W**, **\S** signifiant respectivement **[\^0-9]**, **[\^a-zA-Z0-9_]** et **[\^ \r\t\n\f]**.

6.2.2 Les multiplicateurs de sélecteurs

Des multiplicateurs peuvent être utilisés pour spécifier les expressions régulières :

- * indique 0 ou plus de caractères identiques à celui placé à gauche ;
- + indique 1 ou plus de caractères identiques à celui placé à gauche ;
- ? indique 0 ou 1 caractère identique à celui placé à gauche.

chaîne source	expr. reg	V/F	contenu de \$&
=====	=====	=====	=====
'xxxF'	x?	V	x
'xxxF'	x*	V	xxx
"n'importe quoi"	.*	V	"n'importe quoi"
'abcdefg'	a.+d	V	'abcd'
'aabbcdde'	a+b+	V	'aabb'
'aabbcdde'	a+c	F	

Il faut manipuler avec précaution le multiplicateur * car il signifie 0 ou n occurrences (0 inclus évidemment) et de plus, comme le multiplicateur +, il est glouton, c'est à dire que s'il y a correspondance, il retournera la plus longue chaîne possible.

chaîne source	expr. reg	V/F	contenu de \$&
=====	=====	=====	=====
'xxxF'	x*	V	xxx
'abcxxxF'	x*	V	chaîne vide
'abcxxxF'	abcx*	V	abcxxx

Dans l'exemple ci-dessus, la chaîne est parcourue de la gauche vers la droite, dans le premier cas l'opérateur glouton retourne xxx, dans le second cas il retourne 0x (donc une chaîne vide), le fonctionnement du 3ème cas est analogue au premier.

Pour inhiber le fonctionnement glouton d'un multiplicateur, on le fait suivre d'un `?` qui signifie : le nombre minimum de caractères correspondant au sélecteur.

chaîne source	expr. reg	V/F	contenu de \$&
=====	=====	=====	=====
'xxalloallo '	a.*o	V	'alloallo'
'xxalloallo '	a.*?o	V	'allo'
'xyzaaaxyz'	xyza*	V	'xyzaaa'
'xyzaaaxyz'	xyza*?	V	'xyz'

6.2.3 Mise en mémoire d'une sélection partielle

Il est possible dans une même expression régulière d'utiliser comme sélecteur le résultat d'une sélection déjà effectuée, elle est mise en mémoire et numérotée au préalable par l'opérateur `()`. L'opérateur `\numéro` sert à référencer une mise en mémoire :

```
if (/(.*)>.*<\/\1>/) printf ("balise : $& \n");
```

L'exemple ci-dessus¹ peut servir à rechercher certaines balises d'un texte HTML. Les parenthèses autour du premier `.*` indiquent qu'il faut stocker le résultat de la sélection, le `\1` fait référence à cette sélection.

```
'<H1> xxxx </H1>' ou '<EM> xxxxxx </EM>'
correspondent à l'expression régulière
donnée ci-dessus, mais
'<A HREF="xxxx"> ancre </A>' ne le fait pas !
```

¹ recherche n'importe quelle chaîne de caractères située entre `<>` suivie d'un texte (éventuellement vide), suivi entre `<>` de la chaîne trouvée initialement précédée d'un `/`

L'opérateur () est également utilisé pour appliquer un multiplicateur au résultat d'une sélection :

```
if (/ab(cde)+/) {
  # vrai pour abcde, abcdecde, abcdecdecde, ...
  # le multiplicateur + s'applique au résultat de
  # la sélection entre parenthèses.
}
```

6.2.4 La sélection alternative

L'opérateur | est utilisé pour marquer une sélection avec alternative :

```
/abc|def/ sélectionne des chaînes de caractères contenant
consécutivement les caractères abc ou def
```

6.2.5 Balisage des frontières de sélection

Des opérateurs particuliers permettent de spécifier des frontières auxquelles s'appliquent les sélecteurs :

- \b indique que le sélecteur précédent ne s'applique que sur une frontière de mot ;
- \B indique que le sélecteur précédent ne s'applique pas sur une frontière de mot ;
- ^ marque le début d'une ligne ;
- \$ marque la fin d'une ligne.

```
^http      <==> chaîne commençant par la séquence http
[^http]    <==> chaîne ne comportant ni h, ni t, ni t, ni p
\^http     <==> chaîne contenant la suite de caractères
             ^http
xxx$       <==> chaîne terminant par xxx
xxx\$      <==> chaîne contenant la suite de caractères
             xxx$
\bpay\b    <==> chaîne contenant le mot pays, paysage ou
             paysan ne correspondent pas
```

6.2.6 sélection d'un nombre exact d'occurrences

Il est possible de sélectionner un nombre exact d'occurrences en utilisant l'opérateur `{}` :

<code>/\d{1,5}/</code>	recherche un nombre composé de 1 à 5 chiffres (au moins 1 et au plus 5)
<code>/x{3}/</code>	recherche exactement 3 x
<code>/x{3,}</code>	recherche une suite d'au moins 3 x (la borne supérieure n'est pas précisée)

6.2.7 Priorité des opérateurs de construction d'expressions régulières

Comme pour rédaction d'expressions arithmétiques, il peut y avoir des ambiguïtés dans l'écriture des expressions régulières :

<code>[a-z] [A-Z]+</code>	veut il dire 1 ou n occurrences d'une lettre minuscule ou d'une lettre majuscule ou plutôt une lettre minuscule ou 1 ou n occurrences d'une lettre majuscule
---------------------------	--

Des règles de priorités existent, en en tenant compte dans l'exemple ci-dessus, on a programmé le deuxième cas (le `|` est moins prioritaire que le `+`). Pour éviter de connaître précisément les règles de priorités des opérateurs, il suffit de noter que l'opérateur `()` est le plus prioritaire, pour programmer à coup sûr le cas numéro 1, on pourrait écrire :

<code>([a-z] [A-Z])+</code>	<code><==></code> la mise entre parenthèses de l'alternative assure qu'elle sera prioritaire sur le <code>+</code>
-----------------------------	---

6.3 L'opérateur de substitution

Pour effectuer des substitutions correspondant à des expressions régulières, on utilise l'opérateur **s/expr-reg/chaîne/**, qui par défaut utilise et affecte la variable `$_`.

```
$_ = '123456789';
s/123/abc/;           <==> $_ = 'abc456789'
$var = 'abctototodef';
$var =~ s/(to)+//;   <==> $var = 'abcdef'
```

Les variables `$&`, `$'` et `$'` conservent le même sens qu'avec l'opérateur de recherche, la chaîne supprimée est contenue dans `$&` ...

Lorsque la chaîne à traiter n'est pas contenue dans la variable `$_`, on utilise l'opérateur de recherche `=~` qui combiné avec l'opérateur de substitution, utilise et affecte la variable citée.

L'opérateur de substitution s'arrête sur la première occurrence de la chaîne correspondant à l'expression régulière donnée. Pour effectuer une substitution de toutes les occurrences, il convient de l'indiquer en rajoutant un **g** (global) derrière le dernier `/` de l'opérateur de substitution :

```
$var = 'aa123aaa456aaaa789';
$var =~ s/a+/ /g;       <==> $var = ' 123 456 789'
```

Il peut être intéressant de référencer la chaîne correspondant à une expression régulière dans une opération de substitution :

```
while ($ligne = <STDIN>)
{ $ligne =~ s/(images|applet)/$1\monprojet/g; }
```

L'exemple précédent peut servir à rajouter un répertoire (monprojet ici) dans les références à des images ou des applets à l'intérieur d'un document HTML.

6.4 Ignorer la casse

Il est possible d'indiquer aux opérateurs de recherche et de substitution d'ignorer la casse (de traiter indifféremment les lettres majuscules et minuscules) en utilisant le symbole **i** (ignore) :

```
s/begin/debut/i    remplace begin, Begin, BEGIN .... par
                  la chaine "debut"
```

6.5 Travailler sur les champs d'une ligne

6.5.1 split

Perl traite les flux de données comme une suite de lignes composées de champs séparés par les séparateurs habituels (espace, tabulation).

La fonction **split** découpe une ligne en champs en utilisant comme délimiteur une expression régulière, toutes les parties de la ligne qui ne correspondent pas à l'expression régulière sont stockées dans une liste.

```
#!/usr/local/bin/perl

while ($ligne=<STDIN>)
{
    @champ = split (/:/, $ligne);
    printf ("User: %s Sh: %s \n", $champ[0], $champ[$#champ]);
}
```

Le programme ci-dessus examine le contenu d'un fichier style */etc/passwd* (dont les champs sont séparés par le caractère :) et isole les noms d'utilisateurs (le premier champ) ainsi que leurs interpréteurs de commandes favoris (le dernier champ, sous Unix bien sûr).

split travaille implicitement sur la variable `$_`, et on aurait donc pu écrire l'exemple

précédent sous la forme suivante :

```
#!/usr/local/bin/perl

while (<STDIN>)
{
  @champ = split (/:/);
  printf ("User: %s Sh: %s \n", $champ[0], $champ[$#champ]);
}
```

S'il s'agit d'utiliser les espaces ou les tabulations comme délimiteur, on peut tout simplement omettre l'expression régulière. Si on travaille sur le variable `$_`, on peut alors écrire :

```
while (<STDIN>)
{
  @champ = split();    # ou même @champ=split;
}
```

6.5.2 join

join permet l'opération inverse d'un **split** ie. recoller les morceaux.

```
$ligne = join(':', @champ); <==> la variable $ligne reçoit
    la liste @champ avec insertion d'un séparateur (: ici).
```


Chapitre 7

Quelques trucs utiles

7.1 Exécuter des commandes

Une commande placée entre accents graves est soumise par Perl à l'interpréteur de commande du système, le résultat peut facilement être récupéré dans une variable :

```
@who = `usr/bin/who`;  
foreach (@who) {  
    @User=split(/\s+/);  
    printf ("User : %s \n", $User[0]);  
}
```

L'exemple ci-dessus exécute la commande Unix *who* et affiche le nom des usagers (le premier champ des lignes retournées par *who*). C'est la sortie STDOUT qui est redirigée dans la variable spécifiée. Pour rediriger également les erreurs éventuelles, on pourrait écrire (sous Unix) :

```
$rm = `bin/rm fichier 2>&1`;  
if ($rm) { printf ("Erreur ---> %s \n", $rm);}
```

7.2 La fonction die

La fonction **die** arrête l'exécution d'un programme en affichant une liste passée en argument.

```
die ("c'est fini !\n");
```

On l'utilise généralement avec l'instruction de contrôle `||` pour terminer l'exécution d'un programme si l'instruction précédente se passe mal (ouvertures de fichiers, ...) ou si un problème survient dans la logique de déroulement :

```
($chaine) || die ("La chaîne à traiter est vide \n");
```

En indiquant `#!` dans les paramètres passés à la fonction **die**, on affiche le code retour des fonctions systèmes :

```
open (FIC,"MonFichier") || die ("Pb ouverture : $! \n");
```

7.3 Évaluation à la volée

Perl permet d'évaluer à l'exécution des variables contenant du code. Cette fonctionnalité permet de construire dynamiquement des programmes ou morceaux de programmes. On utilise pour cela l'instruction **eval** qui va passer à Perl une chaîne à évaluer :

```
$var = '$som = $val1 + $val2';  
eval $var; # <==> à l'exécution, le contenu de $var est  
passé à Perl qui exécutera son contenu
```

Il convient d'être prudent en utilisant **eval**, ne pas demander à l'utilisateur de rentrer une chaîne de caractères et la faire ensuite évaluer ainsi, il pourrait par exemple avoir accès à l'interpréteur de commandes et saisir `/bin/rm *` ' ...

Perl traite les instructions trouvées dans une variable passée à **eval** comme un bloc (ie. comme si elles étaient délimitées par `{}`) pouvant contenir des variables locales (cf. section 8).

L'instruction **eval** est également utilisée pour traiter les exceptions d'exécution d'un programme Perl, dans ce cas le contenu de la variable à évaluer doit explicitement être écrit comme un bloc complet (avec les accolades de début et de fin).

7.4 Les arguments de la ligne de commande

Les arguments de la ligne de commande sont accessibles par l'intermédiaire du tableau **@ARGV** qui contient un élément par arguments passés au programme. Contrairement au langage C, le premier élément du tableau **@ARGV** ne contient pas le nom de la commande, **\$ARGV[0]** contient le premier argument passé à la commande.

```
foreach (@ARGV) {  
    printf ("Arg = %s \n",$_);  
}
```


Chapitre 8

La portée des variables

Par défaut une variable Perl est globale, elle est donc *visible* dans l'ensemble du programme. De plus Perl n'oblige pas à déclarer les variables avant de les utiliser. Ces deux propriétés sont sources de bien des problèmes de mise au point, Perl permet donc de déclarer des variables locales et d'utiliser un mécanisme obligeant le programmeur à déclarer les variables avant de les utiliser.

Les programmeurs utilisant les langages de programmation les plus courants sont en général habitués à un fonctionnement différent : les variables sont automatiquement locales aux blocs de déclarations, pour les rendre globales, il est nécessaire de le préciser.

8.1 Déclarer des variables locales avec `my`

Il est possible de rendre locale une variable en précédant sa déclaration par **my**, elle ne sera alors connue que du bloc ou de la fonction (cf. section 9) qui contient sa déclaration. Les variables locales sont par défaut initialisées à **undef**¹ et, il n'existe pas de variables locales statiques (celles dont la visibilité est limitée au bloc de déclaration mais dont la valeur est remanente d'une entrée dans le bloc à une autre).

1. On peut les comparer aux variables automatiques de C.

```
#!/usr/local/bin/perl
$I = 10;
{
  my $I = 2;
  {
    $I++;
    {
      my $I = 4;
      printf ("I = %d \n",$I); <==> Affiche I = 4
    }
    printf ("I = %d \n",$I); <==> Affiche I = 3
  }
  printf ("I = %d \n",$I); <==> Affiche I = 3
}
printf ("I = %d \n",$I); <==> Affiche I = 10
```

8.2 Une autre manière de déclarer des variables locales

Les variables locales déclarées avec **my** ne sont visibles que dans leurs blocs de déclarations. Perl propose une autre variété de variables locales : celles qui en plus ont la particularité d'être visibles dans toutes les fonctions (cf. section 9) appelées depuis leurs blocs de déclarations.

```
#!/usr/local/bin/perl

$I = 4; <==> cet I est global
{
  local $I = 2;
  fonc(); <==> Affiche I = 2
}
{
  my $I = 3;
  fonc(); <==> Affiche I = 4
}
sub fonc {printf("I = %d \n", $I);}
```

Les variables déclarées avec **local** ne contribuent pas à la lisibilité d'un code, elles peuvent même être source de confusions, il convient de ne pas les utiliser.

8.3 use strict

strict est un module de la bibliothèque standard de Perl, il permet de générer une erreur à la compilation si une variable accédée n'a pas été préalablement déclarée avec **my** (ou complètement qualifiée, ou importée cf. section 14).

```
#!/usr/local/bin/perl

my $Compteur=10;

$compteur++;          # erreur de frappe
print ('$Compteur = ', "$Compteur \n");

#     La valeur affichée est 10, le programmeur
#     peut ne pas s'en apercevoir.
```

```
#!/usr/local/bin/perl
use strict;

my $Compteur=10;

$compteur++;          # erreur de frappe
print ('$Compteur = ', "$Compteur \n");

# Un message d'erreur est généré à la compilation :
# Global symbol "$compteur" requires explicit package name
```

Dans l'exemple ci-dessus l'erreur de saisie est signalée car Perl n'autorise plus (**use strict**) les déclarations implicites de variables globales. Toutes les variables doivent être déclarées à l'aide de **my** (elles sont locales au programme principal) ou déclarées dans un module externe (cf. section 14). L'utilisation de **use strict** est donc très fortement recommandée.

Chapitre 9

Les fonctions

9.1 Définition d'une fonction

Perl permet d'écrire des fonctions qui admettent ou pas des arguments et qui rendent ou pas une valeur de retour. Elles sont définies par le mot-clef **sub** suivi d'un identificateur et d'un bloc qui va contenir le code de la fonction.

```
sub Ma_Fonction {  
    instruction 1;  
    instruction 2;  
}
```

Les fonctions peuvent être placées n'importe où dans un source Perl (elles sont sautées à l'exécution), il est toutefois conseillé de les placer en début ou en fin de programme. La portée des fonctions est globale, il est possible d'enliser une fonction dans une autre mais cela ne restreint pas sa visibilité, cette forme d'écriture n'a donc pas de sens.

```
sub Ma_Fonction {  
    instruction 1;  
    instruction 2;  
sub Fonction_Enlisee { ... } <==> Fonction_Enlisee est  
    tout de même accessible en dehors de Ma_Fonction  
}
```

Si une fonction retourne une valeur elle est précisée par l'instruction **return** suivi d'une expression à évaluer. Les arguments d'une fonction sont passés par valeurs et récupérés dans le tableau @_ (arobas souligné), en fonction du cas posé diverses solutions sont possibles pour y avoir accès :

```
my ($var1,$var2,$var3) = @_;
# ou
my $var1 = $_[0];
my $var2 = $_[1];
my $var3 = $_[2];
# ou encore
foreach (@_) { .... }
```

Dans les faits une fonction retourne toujours une valeur, c'est celle de la dernière expression évaluée avant la sortie. On peut donc se passer de l'instruction **return** mais sa présence ne nuit pas à lisibilité du code !

9.2 Appel d'une fonction

On appelle une fonction en indiquant son identificateur suivi entre parenthèses des arguments (éventuellement 0). L'appel à une fonction peut être placé dans une expression en fonction de sa valeur de retour :

```
#!/usr/local/bin/perl
$I = 10;
$J = 20;
resultat(); <==> appel d'une fonction qui ne retourne pas
                  de valeur (ou plus exactement dont la
                  valeur de retour est ignorée ici)

sub resultat {
    printf ("La somme est : %d \n",som($I,$J));
}
sub som {
    my ($a,$b) = @_;
    return $a + $b;
}
```

9.3 Prototyper les fonctions

Perl permet de vérifier (à la compilation) que les arguments passés à une fonction correspondent bien à ce qui est prévu par le programmeur. On utilise pour cela le prototypage des fonctions utilisées (qui doivent être *visibles* à la compilation). Le prototypage consiste à indiquer le nombre d'arguments attendus en précisant leur type. En Perl le prototype suit la déclaration du nom de la fonction (entre parenthèses) et utilise les conventions suivantes :

- un `$` indique la présence d'un scalaire;
- un `@` indique la présence d'une liste;
- un `%` indique la présence d'un tableau associatif;
- un `;` sépare les arguments obligatoires des arguments optionnels;
- lorsque le caractère `\` précède un `@` ou un `%` il rend obligatoire que l'argument correspondant soit effectivement une liste ou un tableau associatif (cf. l'exemple suivant).

Il faut faire attention à l'utilisation des `@` et `%` dans les prototypes car ils *mangent* les caractères suivants et forcent un contexte de liste. L'utilisation du `\` devant ces caractères est donc très importante (cf l'exemple suivant).

```
#!/usr/bin/perl

# prototypes des fonctions utilisées
sub fonc1($$$); # 3 scalaires
sub fonc2($$@); # 2 scalaires et une liste
sub fonc3($$\@); # 2 scalaires et une variable de type liste

@liste = (1, 2, 3);
print (fonc1(@liste,5,6)," ",fonc2(5,6,7), fonc3(5,6,7));
#
# ici l'appel de fonc1 rendra 14 car @liste dans un contexte
# scalaire vaut 3 (nombre d'éléments). L'appel de fonc2
# rendra 18 car 7 est pris comme une liste de 1 élément.
# En l'état ce programme ne compile pas car l'appel à fonc3
# est incorrect, le 3ème argument n'est pas une variable de
# type liste.
#

sub fonc1($$$) {
#####
    my ($a, $b, $c) = @_;
    return($a+$b+$c);
}
sub fonc2($$@) {
#####
    my ($a, $b, @liste) = @_;
    my $res;
    foreach (@liste) {
        $res += $_;
    }
    return($a+$b+$res);
}
sub fonc3($$\@) {
#####
    my ($a, $b, @liste) = @_;
    my $res;
    foreach (@liste) {
        $res += $_;
    }
    return($a+$b+$res);
}
```

9.4 Passer des arguments par références

Les arguments d'une fonction sont récupérés dans le tableau @_ , ce mécanisme est suffisant pour passer quelques valeurs scalaires, mais devient très difficile à gérer s'il est nécessaire de passer plusieurs tableaux en paramètres (comment les délimiter entre eux ?). Pour résoudre ce problème, Perl propose d'utiliser des références plutôt que des valeurs.

9.4.1 Les références de variables

On référence une variable Perl en précédant son identificateur d'un \ :

```
$scalaire = 4;
@tableau = (1,2,3,4,5,6);

$refscal = \$scalaire;
$reftab = \@tableau;
$refhac = \%hache;
```

Une référence est un variable de type scalaire, on peut donc corriger ce qui a été indiqué en section 2.1, un scalaire est un nombre, une chaîne de caractère ou **une référence**.

Pour déréférencer une variable contenant une référence, on précède son identificateur d'un \$, @ ou % en fonction du type de données qu'elle référence.

```
$scalaire = 4;
@tableau = (1,2,3,4,5,6);
%hache = ("a",1,"b",2,"c",3);

$refscal = \$scalaire;
$reftab = \@tableau;
$refhac = \%hache;

$$refscal = 5;           <==> $scalaire = 5;
@$reftab = (7,8,9)      <==> @tableau = (7,8,9);
$$reftab[0] = 2;        <==> $tableau[0] = 2;
%$refhac = ("d",4,"e",5,"f",6);
                        <==> %hache = ("d",4,"e",5,"f",6);
```

Pour déréférencer une variable contenant une référence sur un tableau ou un tableau associatif, on dispose d'une notation fléchée qui est plus confortable à utiliser que la notation utilisant le `$$`:

```
@tableau = (1,2,3,4,5,6,7);
%hache = ('bleu',1,'rouge',2,'vert',3);
$reftab = \@tableau;
$refhache = \%hache;

$reftab->[1] = 10; <==> $$reftab[1] = 10;
$refhache->{'rouge'} = 4; <==> $$refhache{'rouge'} = 4;
```

Il est également possible de créer des références des tableaux anonymes ou des hashes anonymes. Cette forme d'écriture est utilisée pour manipuler des structures de données complexes (tableaux de tableaux, tableaux de hashes ... cf. section 12).

Une référence sur un tableau anonyme se crée en utilisant des crochets (`[]`):

```
@tableau = (1,2,3,4);
$reftab = \@tableau;
# peut aussi s'écrire
$reftab = [1,2,3,4];
# l'accès aux éléments est inchangé
$reftab->[1] ou $$reftab[1]
```

Une référence sur un hash anonyme se crée en utilisant des accolades (`{}`):

```
%hache = ('bleu',1,'rouge',2,'vert',3);
$refhache = \%hache;
# peut aussi s'écrire
$refhache={'bleu',1,'rouge',2,'vert',3};
# l'accès aux éléments est inchangé
$refhache->{'rouge'} = 4;
```

Les références anonymes en Perl sont très pratiques à utiliser car le programmeur (contrairement au langage C par exemple) n'a pas à se soucier de la gestion de la mémoire (taille demandée et libération). Perl gère un compteur des références à chaque valeurs, qu'elles soient référencées directement ou pas, celles qui ne sont plus référencées sont détruites automatiquement.

9.4.2 Un exemple

Dans l'exemple suivant on désire réaliser une fonction recevant deux listes en arguments (@tab1 et @tab2), elle va rendre une liste résultante contenant la première liste triée suivie de la seconde également triée. On utilise ici un passage des arguments par références car on utilise deux tableaux, par valeurs la variable @_ ne pourrait pas nous renseigner sur la limite du tableau @tab1. Par références on peut passer un nombre très important de tableaux car les références sont des scalaires.

```
#!/usr/local/bin/perl

@tab1 = (7,3,2,8,9,1,2);
@tab2 = (3,2,1,4);
# Appel de la fonction tri avec deux arguments
# qui sont des références sur les tableaux.
@ltri = tri(\@tab1, \@tab2);
# Affichage du résultat.
print (@ltri);

# Fonction tri
sub tri {
#####
# Récupération des arguments dans 2 variables
# locales qui contiennent des références sur
# les tableaux @tab1 et @tab2.
my ($reftab1, $reftab2) = @_;
my (@tril, @tri2);
# Tri de chaque tableau, on déréférence pour
# pouvoir appeler la fonction sort().
@tril=sort(@$reftab1);
@tri2=sort(@$reftab2);
# Retour du résultat
push(@tril,@tri2);
return @tril; # retour de @tril (par valeur)
}
```


Chapitre 10

Accès au contenu des fichiers

En Perl, l'accès au contenu des fichiers s'effectue à travers des descripteurs de fichiers qui font le lien entre le programme et son environnement extérieur. Comme vu en section 4 tout programme Perl hérite de 3 descripteurs positionnés par le système d'exploitation : STDIN (entrée standard), STDOUT (sortie standard) et STDERR (erreurs produites par les appels au système).

10.1 Ouverture

Pour utiliser un fichier (autre que STDIN, STDOUT ou STDERR) il faut d'abord l'ouvrir (effectuer l'association entre un descripteur de fichier et le nom externe du fichier considéré). On utilise l'instruction **open** suivie d'un descripteur de fichier et du nom externe du fichier ciblé (éventuellement précédé du mode d'ouverture):

```
open (FIC1,'monfichier');      # ouverture en lecture
open (FIC2,'c:\\tmp\\truc');  #
open (FIC3,'>toto');          # > pour ouverture en ecriture
open (FIC3,'>>tutu');         # >> pour ouverture en ajout
open (FIC4,'+<tata');         # +< pour ouverture en lect/ecr
```

`open` retourne vrai si l'ouverture se passe bien, il est important de le vérifier. On peut utiliser un **die** conditionnel pour traiter les valeurs de retour de `open` :

```
open(FIC,'MonFichier') || die("Pb d'ouverture\n");
```

10.2 Lecture

L'opérateur `<>` permet de lire une ligne¹ dans le fichier désigné par le descripteur ciblé. Il rend vrai tant que la fin du fichier n'est pas atteinte :

```
open (FIC,'MonFic') || die ("Le fichier n'existe pas\n");
while (<FIC>)
{
  # par défaut chaque ligne lue (y compris le délimiteur
  # de fin de ligne) est stockée dans $_
}
```

L'opérateur `<>` est adapté à la lecture des fichiers de textes, pour lire des fichiers formatés différemment, on peut utiliser la fonction `sysread()` qui permet de préciser le nombre de caractères à lire ainsi que la variable scalaire qui va contenir les caractères lus. La fonction `sysread()` rend le nombre de caractères effectivement lus :

```
open (FIC,'MonFic') || die ("Le fichier n'existe pas\n");
while (($nb=sysread(FIC,$enr,100) != 0)
{
  # on a lu $nb caractères (100 maxi) dans $enr
}
```

1. On peut aussi lire entièrement un fichier dans une variable de type tableau (cf. section 4).

10.3 Ecriture

print et **printf** permettent d'écrire des lignes dans un fichier. **printf** (cf. section 4.2) permet de préciser le format des variables à écrire, **print** est plus rudimentaire.

```
print FIC ($a,':',$b,':',$c,"\n");  
printf FIC ("%0.5s:%03d:%010s\n",$a,$b,$c);
```

print retourne **1** si l'écriture s'est bien déroulée et **0** en cas d'erreur.

Lorsque l'on écrit un fichier en Perl, il faut se poser la question de sa lecture ultérieure, et notamment celle de la délimitation des champs. Dans l'exemple ci-dessus, il sera facile d'utiliser **split** (cf. 6.5.1) pour récupérer les variables².

Comme pour la lecture, il est possible d'écrire dans un fichier en précisant le nombre de caractères à écrire, c'est le rôle de la fonction **syswrite()**:

```
$nb=syswrite(FIC,$enr,100);  
# écriture du contenu de la variable $enr à concurrence  
# de 100 caractères maximum. $nb contient le nombre de  
# caractères effectivement écrits.
```

10.4 Fermeture

On ferme un fichier en précisant son descripteur à l'aide de la fonction **close**.

```
close(FIC);
```

2. (\$a,\$b,\$c) = split (/:/);

Chapitre 11

Manipulations du système de gestion de fichiers

11.1 Accès aux répertoires

Perl permet d'accéder au contenu des répertoires d'un système de fichiers en utilisant la *convention* *, ou en appelant des fonctions qui procurent une interface d'accès à leur structure.

11.1.1 Utiliser la convention *

Une expression placée entre les symboles < et > va solliciter le système de gestion de fichiers et obtenir une liste des fichiers ou répertoires qui lui correspondent (dans le répertoire de travail ou dans un répertoire cité par l'expression elle même) :

```
# accès aux programme c d'un répertoire
@progc = <*.c>;
# affichages du nom des fichiers sélectionnés
foreach (@progc) {printf ("%s \n", $_);}
```

Utilisée dans un contexte scalaire la convention * va parcourir le répertoire ciblé et rendre successivement les fichiers qui correspondent (plus **undef** quand il n'y a plus de corres-

pondance). On pourrait donc écrire l'exemple précédent de la façon suivante :

```
while (<*.c>) {printf("%s \n",$_);}

```

L'utilisation de la convention * peut s'effectuer en remplaçant les symboles < et > par la fonction **glob** :

```
@include=</usr/local/include/*.h>;
peut aussi s'écrire
@include=glob('/usr/local/include/*.h');

```

11.1.2 Utiliser une interface d'accès

Il est possible de parcourir un répertoire en utilisant la fonction **readdir** qui fonctionne à partir d'un descripteur de répertoire préalablement ouvert par **opendir**. Utilisée dans un contexte scalaire, **readdir** rend successivement le noms des fichiers du répertoire (plus **undef** quand il n'y en a plus). Utilisée dans un contexte de liste, **readdir** rend la liste de tous les fichiers du répertoire ciblé. La fonction **closedir** ferme un descripteur de répertoire ouvert par **opendir**.

```
opendir (DIR, '.') || die ('Erreur Open Dir');
@fic = readdir(DIR);
foreach (@fic) {printf ("%s \n",$_);}

```

11.2 Manipulation des fichiers et répertoires

Perl procure des fonctions pour réaliser les actions les plus usuelles sur les fichiers et répertoires.

11.2.1 Changer de répertoire de travail

La fonction **chdir** permet de changer le répertoire de travail de l'application en cours.

```
chdir('c:\\windows\\system') || die ("Erreur chdir \n");
```

11.2.2 Créer un répertoire

La fonction **mkdir** permet de créer un répertoire en positionnant des droits d'accès à la mode Unix¹

```
mkdir ('MonRepert',0755) || die ("Err. Cr. répertoire \n");
```

11.2.3 Supprimer un répertoire

Un répertoire vide peut être supprimé par la fonction **rmdir**.

```
rmdir ('MonRepert') || die ("Err. Sup. répertoire \n");
```

11.2.4 Supprimer un fichier

La fonction **unlink** permet de supprimer une liste de fichiers.

```
foreach (<*.old>) {  
    unlink($_); || die ("Erreur suppression \n");  
}
```

1. Une équivalence est faite automatiquement pour Win32.

11.2.5 Renommer un fichier

La fonction **rename** permet de changer le nom d'un fichier.

```
rename("log","log-old"); || die ("Pb. logs \n");
```

11.2.6 Modifier les droits d'accès

La fonction **chmod** permet de positionner les droits d'accès à un fichier à la manière d'Unix² en utilisant un codage octal³.

```
chmod(0755,$fic); || die ("Err. droits d'accès \n");
```

11.3 Fonctions utiles

Il est possible d'obtenir des informations sur un fichier ou un répertoire en utilisant des expressions dont l'écriture est dérivée de celle de la commande Unix **test**. Ces expressions s'écrivent à l'aide d'un opérateur suivi d'une chaîne de caractères⁴ qui représente le nom potentiel d'un fichier ou répertoire⁵. Les opérateurs (sauf exceptions) rendent une valeur vraie ou fausse, les principaux sont les suivants :

-r	fichier ou répertoire accessible en lecture
-w	fichier ou répertoire accessible en écriture
-e	fichier ou répertoire existant
-x	fichier exécutable
-z	fichier existant mais vide
-s	fichier ou répertoire non vide, la valeur retournée est la taille en octet
-f	fichier normal (répertoire et spéciaux exclus)
-d	répertoire

2. On procède par équivalence pour Win32.

3. La signification du codage est documentée dans le manuel Unix, voir `chmod(1)`.

4. Elles fonctionnent également sur des descripteurs de fichiers.

5. En l'absence de chaîne de caractères et de descripteur de fichier, c'est la variable `$ _` qui sera utilisée.

```
$InstallDir = "MonAppli";  
(-d $InstallDir) || die ("Appli non installée \n");
```

La fonction **stat** permet d'obtenir plus de détails sur un fichier matérialisé par une chaîne de caractères ou un descripteur de fichier. Elle donne accès aux différents champs d'une entrée dans un système de fichiers⁶. On a alors accès à la taille d'un fichier, à sa date de création ... **stat** renvoie une liste de 13 valeurs dans un tableau, la 8ème correspond à la taille du fichier.

L'exemple ci-dessous utilise la fonction **stat** pour déterminer la taille d'un fichier :

```
#!/usr/local/bin/perl  
print ("Entrez un nom de fichier : \n");  
while (<STDIN>)  
{  
  chomp;  
  (-f) ? print (taille($_),"\n") : print ($_," inconnu\n");  
  print ("Entrez un nom de fichier : \n");  
}
```

```
sub taille {  
  my ($fic) = @_;  
  my ($dev,$inode,$perm,$liens,$uid,$gid,  
      $ndev,$lg,$acces,$mod,$scr,$blksize,$bl)=stat($fic);  
  return($lg);  
}
```

6. Une inode sous Unix.

Chapitre 12

Les structures de données complexes

Perl autorise de manipuler des structures de données plus complexes que celles accessibles par les 3 types de base. Des références sur des listes anonymes ou des hashes anonymes sont utilisées pour cela.

12.1 Les listes de listes

```
@Liste = (  
    ['toto', 'tutu', 'tata'],  
    ['truc', 'much'],  
    ['patati', 'patata']  
);
```

Dans l'exemple ci-dessus, **@Liste** est une liste de références sur des tableaux anonymes (I).

L'affectation d'un élément s'opère à la manière d'un tableau à deux dimensions. L'affectation d'une liste complète implique l'utilisation de références.

```
# remplacer 'truc' par 'machin'  
$Liste[1][0]='machin';  
# remplacer ['patati', 'patata'] par ['bla', 'blabla']  
@AutreListe = ('bla', 'blabla');  
$Liste[2] = \@AutreListe;
```

```

# Erreur à ne pas commettre
$Liste[2] = @AutreListe;
# qui charge la valeur 2 (le nombre d'éléments) en lieu
# et place d'une référence sur une liste ...

# Il n'est pas forcément nécessaire de déclarer une liste
# pour assurer l'opération. On utilise ci-dessous une
# référence sur une liste anonyme ([]).

$Liste[2] = ['bla','blabla'];

# ajouter une liste
push(@Liste, ['en','voila','du','texte']);

```

L'accès à un élément s'effectue à la manière d'un tableau à deux dimensions. L'utilisation d'une des listes d'un tableau de listes dans une opération d'affectation nécessite un peu d'attention :

```
@AutreListe = @$Liste[2];
```

Cette écriture semble naturelle, **\$Liste[2]** est une référence sur une liste, pour déréférencer on la précède du caractère @. En fait, pour des raisons de priorités entre opérateurs **\$Liste** sera considéré comme une référence (qui n'existe pas), puis on fera le déréférencement et enfin on cherchera le 2ème élément. L'utilisation de **use strict** (cf. section 8.3) indiquerait d'ailleurs une erreur à la compilation !

Il faut donc écrire :

```
@AutreListe = @{ $Liste[2] }
```

Les { et } sont utilisés pour clairement indiquer que la référence est \$Liste[2].

12.2 Les hashes de hashes

```
%User = (
  toto => {
    Homedir => '/Home/toto',
    Passwd  => 'AlPefjWNa03H2',
    Email   => 'toto@ici.fr'
  },
  tutu => {
    Homedir => '/Home/tutu',
    Passwd  => '31PrfjWNa08Ha',
    Email   => 'tutu@ici.fr'
  },
  tata => {
    Homedir => '/Home/tata',
    Passwd  => '41aqfjer508Ha',
    Email   => 'tata@ici.fr'
  }
);
```

Dans l'exemple ci-dessus, `%User` est un hash de hashes anonymes (`{}`) (à une clef correspond une référence sur un hash anonyme).

```
# Changer une valeur
$User {'tutu'} {'Email'} = 'tutu@parla.fr';

# Ajouter une nouvelle clef
$User {'titi'} = {HomeDir => '/home/titi',
                  Passwd  => '32drtyuoiXes',
                  Email   => 'titi@ici.fr'};

# Accéder à la totalité des éléments
foreach $util (keys %User) {
  foreach (keys %{$User{$util}}) {
    print ("$util->$_ : $User{$util}{$_} \n");
  }
}
```

Dans l'exemple ci-dessus on écrit `%{$User{$util}}` et non pas `;%User{$util}` pour les mêmes raisons qu'indiquées en section 12.1.

12.3 Autres structures de données

La construction de structures de données ne s'arrête pas aux listes de listes ou aux hashes de hashes. Il est possible de généraliser le procédé pour bâtir des structures hybrides mêlant des listes et des hashes sur plusieurs niveaux. Pour bien comprendre le fonctionnement, il faut bien avoir à l'esprit qu'un tableau en Perl est toujours uni-dimensionnel. Un tableau à 4 dimensions ne sera en réalité qu'un tableau à une dimension contenant des références vers des tableaux contenant des références vers des tableaux contenant des références vers des tableaux.

Dans l'exemple suivant, on gère un tableau (à l'allure tri-dimensionnelle) de valeurs correspondant à des années, mois et jours. On commence par installer quelques valeurs dans le tableau avant de le parcourir entièrement pour calculer la somme de toutes les valeurs installées.

```
#!/usr/local/bin/perl

# Exemple d'utilisation d'un tableau à 3 dimensions.

@Valeurs = ();

# Initialisation du tableau

AddValeurs(1999,12,1,150);
AddValeurs(1999,12,1,250);
AddValeurs(1994,12,2,100);
AddValeurs(1999,12,3,500);
AddValeurs(1999,11,3,500);
AddValeurs(1999,11,8,500);
AddValeurs(1990,11,10,500);

$NbElem = @Valeurs;

# $NbElem vaut 2000 [0..1999], car Perl bouche
# les trous.

print ("Total Annuel : ",StatAn(1999),"\n");
```

```

sub AddValeurs {
#####
    my ($an, $mois, $jour, $montant) = @_;

#   On utilise ici une écriture traditionnelle pour
#   un tableau à 3 dimensions.

    $Valeurs [$an] [$mois] [$jour] += $montant;
}

sub StatAn {
#####
    my ($an) = @_;
    my $total;

    foreach $mois (@{$Valeurs[$an]}) {

#   @{$Valeurs[$an]} est une liste de références
#   13 ici ([0..12]) car le plus grand mois cité
#   vaut 12 pour l'année 1999.
#   $mois contient une référence vers un tableau qui
#   contient les valeurs associées. Il y a ici 9
#   valeurs pour le 11ème mois de l'année 1999 et
#   4 valeurs ([0..3]) pour le 12ème mois de 1999.

        foreach $jour (@$mois) {
            $total += $jour;
        }
    }
    return($total);
}

```

On peut maintenant compléter l'exemple de la section 12.2 pour associer à chaque utilisateur une liste de machine sur lesquelles il est autorisé à se connecter. On obtient ainsi un hash de hash de liste.

```
%User = (  
    toto => {  
        Homedir => '/Home/toto',  
        Passwd  => 'A1PefjWNa03H2',  
        Email   => 'toto@ici.fr',  
        Login   => ['mach1','mach2','mach3']  
    },  
    tutu => {  
        Homedir => '/Home/tutu',  
        Passwd  => '31PrfjWNa08Ha',  
        Email   => 'tutu@ici.fr',  
        Login   => ['mach2','mach4','mach3','mach8']  
    }  
);  
  
# pour accéder à la liste des machines d'un utilisateur  
  
@liste = @{$User{'tutu'}{'Login'}};  
  
# $User{'tutu'}{'Login'} est une référence sur une liste,  
# on la place entre @{...} pour la déréférencer.
```

Chapitre 13

Le débogueur de Perl

Perl propose un environnement interactif pour le débogage des programmes. Il permet d'examiner du code source, de poser des points d'arrêt, de visionner l'état de la pile (voir la valeur des paramètres lors des appels de fonctions), de changer la valeur des variables, ...

Le débogueur est appelé en spécifiant l'option **-d** lors de l'appel de Perl, on entre alors dans un mode interactif, le débogueur attend des commandes pour dérouler le code comme dans l'exemple suivant (où le programme comporte au moins une erreur) :

```
1  #!/usr/local/bin/perl
2  my ($moy, $total, $nb);
3  print ("Entrez un nombre \n");
4  while (<STDIN>) {
5      $total += $_;
6      print ("Entrez un nombre \n");
7  }
8  $moy = $total / $nb;
9  print ("la moyenne est : $moy \n");
```

Le programme ci-dessus provoque une division par zéro, pour le mettre au point on appelle le débogueur. Dans l'exemple suivant, le texte précédé par >> est généré par Perl, celui précédé par << est fourni par le programmeur qui effectue la mise au point.

```

perl -d debug.pl
>> Loading DB routines from perl5db.pl version 1.0402
>> Emacs support available.
>> Enter h or 'h h' for help.

>> main::(debug.pl:2):      my ($moy, $total, $nb);
>> DB<1>  << /\$moy/
>> 8:     $moy = $total / $nb;
>> DB<2>  << b 8
>> DB<3>  << c
>> Entrez un nombre
<< 2
>> Entrez un nombre
<< 4
>> Entrez un nombre
<< Ctrl-d
>> main::(debug.pl:8):     $moy = $total / $nb;
>> DB<3>  << x $total
>> 0 6
>> DB<4>  << x $nb
>> 0 undef
>> DB<5>  << $nb=2
>> DB<6>  << c
>> la moyenne est : 3

>> Debugged program terminated. Use q to quit or R
>> to restart,
>> use O inhibit_exit to avoid stopping after
>> program termination,
>> h q, h R or h O to get additional info.
>> DB<6> << q

```

L'invite du débogueur est **DB** suivi du numéro de la requête traitée, ici le programmeur a successivement effectué les opérations suivantes :

- recherche de la prochaine ligne de code exécutable qui contient la chaîne **\$moy** (**/\\$moy/**);
- pose d'un point d'arrêt sur la ligne 8 qui correspond à la recherche effectuée (**b 8**);
- demande l'exécution du programme jusqu'au prochain point d'arrêt (**c**);
- rentre 2 valeurs (2 et 4) et marque la fin de saisie (**Ctrl-d**);
- demande à consulter le contenu de la variable **\$total** lorsque le débogueur s'arrête

- sur le point d'arrêt (**x \$total**);
- demande l'affichage de la variable **\$nb (x \$nb)** et constate qu'elle est la source de l'erreur (**undef**, c'est à dire 0);
- affecte 2 à la variable **\$nb (\$nb=2)**.
- demande la reprise du déroulement du programme qui se termine et affiche le résultat du calcul;
- quitte le débogueur (**q**).

Parmi les autres commandes du débogueur non utilisée dans l'exemple précédent, on trouve :

- S, demander un déroulement pas à pas;
- T, afficher la pile (et voir les paramètres transmis à une fonction);
- t, afficher ou ne plus afficher les lignes de programmes exécutées (c'est une bascule désamorcée par défaut);
- D, supprimer tous les points d'arrêts.

Chapitre 14

Écriture et utilisation de modules

Perl autorise l'écriture de modules qui peuvent être assemblés pour former une application. Cette fonctionnalité a permis le développement de nombreuses contributions, des modules ont été développés pour interfacer TCP, CGI, FTP ... c'est une des raisons du succès de Perl.

14.1 Inclure du code

La forme la plus simple de l'écriture modulaire est l'inclusion de code, autorisée en Perl par la fonction **require** suivie d'une expression (en général une chaîne de caractère représentant un nom de fichier contenant le code à inclure). Il s'agit d'une fonctionnalité équivalente au **#include** du langage C, on regroupe en général les inclusions en début de programme :

```
#!/usr/local/bin/perl
require "cgi-lib.pl";    # Inclusion de la bibliothèque
                        # gérant l'interface CGI
```

Les modules Perl écrits ainsi sont souvent un peu *anciens* et proposent des interfaces rudimentaires :

```
#!/usr/local/bin/perl
#
# Programme principal qui utilise le module monmodule.pl
#
require "monmodule.pl";

# on peut simplement appeler la fonction fonc() définie dans
# monmodule.pl et aussi avoir accès à sa variable $I ...

fonc();
print ("I = $I \n");
```

```
# monmodule.pl
#
# Petit module accessible simplement, les variables et
# fonctions sont directement accessibles ...

$I = 2;
sub fonc {
    print ("\appel de fonc() \n");
}
```

Perl cherche les modules à inclure dans le répertoire courant et dans les répertoires cités dans le tableau `@INC`. Pour ajouter des répertoires au tableau `@INC` il convient de modifier la variable d'environnement `Perl5LIB` ou de modifier `@INC` avant le **require** :

```
#!/usr/local/bin/perl
unshift(@INC, "/usr/local/perl/lib/maison");
require "cgi-lib.pl"; # Inclusion de la bibliothèque
                     # gérant l'interface CGI
```

Les fichiers inclus par **require** sont chargés à l'exécution du programme. Il en résulte une souplesse d'usage liée au fait que l'on peut inclure des fichiers dynamiquement (en fonction du contenu d'une variable). On n'est toutefois pas sûr lorsqu'un programme démarre que tout est prêt pour son bon déroulement (un fichier à inclure qui n'existe pas, qui comporte une erreur de syntaxe ...).

Si l'expression citée par **require** est numérique, il s'agit du numéro de version de Perl requis pour ce qui suit.

14.2 Les packages

En écrivant classiquement des modules destinés à être inclus par **require**, le programmeur doit faire attention à la visibilité des variables (cf. section 8). Une première solution est l'usage de **my**, Perl offre toutefois une autre solution : les **packages** inclus par **use**.

Une partie de code isolée dans une déclaration package est destinée à avoir une visibilité externe organisée, le programmeur peut indiquer explicitement quelles sont les variables, les fonctions qui peuvent être simplement utilisées par les consommateurs.

On peut trouver plusieurs déclarations de packages dans un fichier, ce n'est toutefois pas très utilisé, on place généralement un package dans un fichier dont le suffixe est **.pm** (Perl Module).

```
package essai;      # Le nom du package et du fichier .pm

use Exporter;      # appel au module gérant la visibilité
@ISA=('Exporter'); # hérite d'Exporter (non traité ici,
                  # voir la section sur les objets)

@EXPORT_OK=('Fonc1','Fonc2','$Var');

# Le tableau @EXPORT_OK est utilisé pour préciser les
# identificateurs qui sont visibles de l'extérieur du
# package.

$Var = "visible";
$I = 10;

sub Fonc1 { ... }
sub Fonc2 { ... }
sub Fonc3 { ... }
sub Fonc4 { ... }
```

Dans l'exemple ci-dessus, seulement deux fonctions (*Fonc1* et *Fonc2*) ainsi qu'une variable (*\$Var*) peuvent être utilisées depuis l'extérieur du package.

Pour utiliser le package essai (placé dans le fichier essai.pm) il convient de procéder

comme suit :

```
#!/usr/local/bin/perl

# use est suivi du nom du package et des variables
# et fonctions à importer

use essai ('Fonc1','Fonc2','$Var');

Fonc1();
Fonc2();
print ("$Var \n");
# Un appel à Fonc3 produirait une erreur
```

Il est possible d'utiliser le tableau **@EXPORT** à la place de **@EXPORT_OK**, dans ce cas, même si le programmeur déclare vouloir importer seulement un sous ensemble des identificateurs exportés, il aura une visibilité sur l'ensemble. Il est donc préférable d'utiliser **@EXPORT_OK** qui évite les conflits potentiels.

Les identificateurs exportés constituent l'interface du package avec l'extérieur, PERL procure toutefois une autre interface avec les packages : le nommage explicite. Il est possible d'accéder à un identificateur (exporté ou non) en le précédant du nom de son package d'appartenance :

```
#!/usr/local/bin/perl

use essai;          # il n'y a pas d'obligation a importer

$I = 20;
essai::Fonc1(); # les identificateurs exportés ou non
essai::Fonc2(); # sont accessibles.
essai::Fonc3();
printf ("%d \n",$essai::I); # affiche 10 (le I de essai.pm)
```

La démarche import/export est évidemment beaucoup plus claire et se généralise.

Les packages peuvent être dotés de constructeurs et de destructeurs rédigés sous la forme de fonctions appelées respectivement **BEGIN** et **END**. Les constructeurs sont exécutés

dès que possible (à la compilation) et les destructeurs le plus tard possible (lorsque le programme se termine normalement ou pas).

```
package essai;

BEGIN { # sub n'est pas obligatoire
    print ("debut du package \n");
}

beurk();

END { # sub n'est pas obligatoire
    print ("fin du package \n");
}
```

Le package *essai.pm* de l'exemple ci-dessus produit les sorties suivantes :

```
debut du package
  Undefined subroutine &essai::beurk called
                        at essai.pm line 7.
  BEGIN failed--compilation aborted at ...
fin du package
```

14.3 Remarques

La différence indiquée ici entre **use** et **require** est fictive (elle correspond toutefois à un usage largement répandu), il est en fait parfaitement possible d'inclure un package en utilisant **require**. Dans ce cas la routine d'import n'est pas appelée mais elle peut l'être *à la main*, ceci permet d'inclure des packages dynamiquement :

```
if ($a) {
    require "moda"; # Inclusion de moda.pm ou modb.pm
    moda->import(); # en fonction du contenu d'une variable
}
else {
    # L'import n'est pas effectué (require)
    require "modb"; # on le fait donc explicitement.
    modb->import();
}
```


Chapitre 15

Écriture orientée objet

15.1 Un exemple limité à l'usage classique des packages

Si l'on considère une classe d'objet comme une structure de données accompagnée de procédures qui régissent le comportement de ses membres potentiels, alors en Perl on peut essayer d'utiliser les packages.

Dans l'exemple suivant on désire gérer des *objets* d'une *classe* appelée *voiture* caractérisés chacun par une marque et une couleur, à tout moment on souhaite connaître le nombres de voitures en cours.

```
#!/usr/local/bin/perl
use voiture;

$voit1 = voiture::nouvelle('verte','citroen');
$voit2 = voiture::nouvelle('bleue','renault');
$voit3 = voiture::nouvelle('rouge','citroen');

printf (" %s \n", voiture::couleur($voit1));
printf (" %s \n", voiture::marque($voit2));
printf (" %s \n", voiture::couleur($voit3));
printf (" %d \n", voiture::total());
```

```

package voiture;
BEGIN { @liste=();}

sub nouvelle {
    my ($color,$marque) = @_;
    my $objptr = {};
    $objptr->{'couleur'}=$color;
    $objptr->{'marque'}=$marque;
    push(@liste,$objptr);
    return $objptr;
}
sub couleur {
    my ($objptr) = @_;

    return ($objptr->{'couleur'});
}
sub marque {
    my ($objptr) = @_;

    return ($objptr->{'marque'});
}
sub total {
    return($#liste+1);
}
1

```

Dans l'exemple précédent, la *classe voiture* est un package, le constructeur de package BEGIN est utilisé pour initialiser la liste des *objets* à vide.

La fonction *nouvelle* est le constructeur de la *classe*, elle reçoit en arguments la couleur et la marque d'un *objet* et retourne la référence de l'*objet* instancié. La référence de l'*objet* est ici une référence sur un hash anonyme ($\$objptr = \{\}$ cf. section 9.4.1). Les fonctions *couleur* et *marque* sont les *méthodes d'instance* de la *classe voiture*. Les fonctions *nouvelle* et *total* sont des *méthodes de classe*.

Si le début du code qui utilise la *classe voiture* s'apparente à du code orienté objet (appel au constructeur notamment), l'utilisation des méthodes s'en éloigne. La référence à l'*objet* ($\$voit3$ par exemple) n'a pas de souvenir du type de l'*objet* référencé et on ne peut l'utiliser qu'en citant le nom du package.

Perl propose donc un type de références particulier, il va permettre d'appeler les méthodes directement à partir de la référence à l'*objet*. Dans l'exemple précédent on souhaite plutôt

écrire `$voit3->couleur()` et non pas `voiture::couleur($voit3)`.

15.2 Référencer une classe

Dans l'exemple de la classe *voiture* les références `$voit1`, `$voit2` et `$voit3` n'ont pas connaissance de la classe d'appartenance des objets qu'elles référencent. La fonction **bless** est utilisée pour cela, on peut maintenant modifier le constructeur de la classe *voiture* comme suit :

```
sub nouvelle {
    my ($classe,$color,$marque) = @_; # le nom de la classe
                                     # est le 1er argument

    my $objptr = {};

    $objptr->{'couleur'}=$color;
    $objptr->{'marque'}=$marque;
    bless $objptr;                    # référence la classe
    push(@liste,$objptr);
    return $objptr;
}
```

Un affichage de la valeur de `$objptr` indique explicitement la classe de l'objet référencé (`voiture=HASH(0xca454)`).

Le programme qui utilise la classe *voiture* peut maintenant être modifié de la manière suivante :

```
#!/usr/local/bin/perl
use voiture;

$voit1 = voiture->nouvelle('verte','citroen');
$voit2 = voiture->nouvelle('bleue','renault');
$voit3 = voiture->nouvelle('rouge','citroen');

printf (" %s \n", $voit1->couleur());
printf (" %s \n", $voit2->marque());
printf (" %s \n", $voit3->couleur());
printf (" %d \n", voiture->total());
```

Il convient ici de noter la forme d'appel des méthodes de classe (**voiture->nouvelle()**) et la forme d'appel des méthodes d'instance (**\$voit1->couleur()**). Les méthodes de classe s'appellent en précédant la méthode par le nom de la classe, et, le nom de la classe est alors passé en premier argument (ce qui justifie le changement intervenu dans notre classe *voiture*).

Les méthodes d'instance reçoivent comme premier argument une référence d'objet, il n'est donc pas nécessaire de modifier le code de notre classe *voiture*.

15.3 Hériter d'une classe

Perl autorise de construire de nouvelles classes à partir de classes existantes en utilisant des techniques d'héritages :

- la liste **@ISA** indique à un package (à une classe) quels sont les ancêtres dont il hérite;
- de manière à pouvoir être hérité, le constructeur d'une classe de base doit rendre une référence *blessee* de la classe spécifiée (de la classe dérivée).

Pour construire une classe appelée *location* (gérant la location de véhicules) héritant de notre classe *voiture*, on peut procéder comme suit :

```
package location;
use voiture;          # inclure voiture.pm pour la compile
@ISA = ("voiture"); # hériter des méthodes de la classe
                    # voiture.

sub loc {
    my ($classe, $couleur, $marque, $nbjour, $pu) = @_;
    # appel du constructeur hérité de la classe de base, il
    # rendra un objet de classe location
    my $locptr = $classe->nouvelle($couleur,$marque);
    $locptr->{'pu'}=$pu;
    $locptr->{'nbjour'}=$nbjour;
    return($locptr);
}

sub tarif {
    my ($locptr) = @_;
    return ($locptr->{'pu'} * $locptr->{'nbjour'});
}
```

Il convient de bien noter que la classe *voiture* doit être modifiée pour que le constructeur *nouvelle()* rende un objet de la classe *location* et non pas un objet de la classe *voiture* :

```
sub nouvelle {
    my ($classe,$color,$marque) = @_; # le nom de la classe
                                       # est le 1er argument

    my $objptr = {};

    $objptr->{'couleur'}=$color;
    $objptr->{'marque'}=$marque;
    bless $objptr,$classe; <=====
    push(@liste,$objptr);
    return $objptr;
}
```

Un programme utilisant la classe *location* peut s'utiliser comme dans l'exemple ci-dessous :

```
#!/usr/local/bin/perl

use location;

$voit1 = location->loc('verte','peugeot',10,100);
print $voit1;
print $voit1->couleur();
print $voit1->tarif();
```


Chapitre 16

L'écriture de scripts CGI

16.1 Introduction

Common Gateway Interface (CGI¹) est une spécification d'interface permettant d'exécuter des procédures externes depuis un service WWW. CGI spécifie (du côté du serveur) comment doit être initialisée la procédure externe et comment elle doit accéder aux informations à traiter. Les informations transmises à la procédure externe sont :

- des variables qui correspondent aux en-têtes de la requête HTTP qui a déclenché l'appel de la procédure externe (HTTP_REFERER, HTTP_USER_AGENT, HTTP_CONNECTION, HTTP_HOST, ...);
- des variables relatives au corps d'un document transmis ou propres au contexte CGI (REQUEST_METHOD, CONTENT_TYPE, CONTENT_LENGTH, REMOTE_HOST, QUERY_STRING, ...);
- un corps de document éventuel.

CGI indique également que si un corps de document est transmis, il doit être accédé via l'entrée standard de la procédure externe. De plus la procédure externe (le script CGI) doit toujours rendre un résultat via sa sortie standard. La spécification de ces deux mécanismes impose (au serveur HTTP) un mode de création assez lourd et coûteux du script CGI (fork + redirection de l'entrée et de la sortie standard + exec). C'est la raison principale du manque de performance des scripts CGI, à chaque requête, le système d'exploitation est fortement sollicité.

1. En service depuis 1993, CGI n'a jamais porté le status de RFC, une proposition est en cours de réalisation cf. <http://web.golux.com/coar/cgi>

Dans la pratique, malgré son manque de performance, CGI reste très utilisé pour transmettre des informations d'un client WWW vers un service particulier (via un serveur HTTP). C'est notamment ainsi que fonctionnent nombre de formulaires WWW, le synoptique des échanges entre client, serveur et script CGI est alors le suivant :

- un client WWW affiche un formulaire ;
- l'utilisateur complète les champs à transmettre et valide le contenu qui est transmis. Si la méthode HTTP utilisée pour transmettre le formulaire est **GET**, alors les champs (l'information à transmettre) sont transmis dans l'URL d'appel du script CGI et lui sont passés via la variable `QUERY_STRING`.
Si la méthode HTTP est **POST**, alors les champs sont transmis dans un corps de document (dont le `CONTENT_TYPE` est **application/x-www-form-urlencoded**) et sont passés au script CGI via son entrée standard. Cette dernière façon de procéder est fortement conseillée.
Le type de document **application/x-www-form-urlencoded** implique que les champs du formulaire sont séparés entre eux par le caractère `&`, qu'ils sont précédés de leur nom (attribut `NAME` de la balise HTML `<FORM>`) et du signe `=`, que les caractères accentués et les espaces sont encodés comme spécifié dans le RFC1738 sur les URL;
- un serveur HTTP réceptionne la requête et initialise le script CGI référencé par le formulaire WWW ;
- le script CGI effectue le traitement demandé et se charge d'émettre un résultat à destination du client WWW instigateur de la requête.

En dehors de toute aide extérieure, un programmeur de script CGI doit donc décoder le **application/x-www-form-urlencoded** pour accéder individuellement aux champs transmis, il doit également prendre en charge l'émission du résultat (et donc générer des entêtes HTTP correctes suivies d'un corps de document HTML). En Perl diverses contributions permettent de faciliter ces deux tâches, les plus connues sont **cgi-lib.pl** et **CGI.pm**.

16.2 Un exemple

Dans l'exemple suivant on désire compléter un formulaire WWW comportant deux champs (nom et prénom), le transmettre à une procédure CGI qui pourrait stocker les champs transmis dans un fichier ... mais se contentera ici d'avertir l'utilisateur de la prise en compte de l'information. Le code HTML du formulaire est placé dans un document à part (extérieur à la procédure CGI²), on vérifie que la méthode HTTP utilisée est bien **POST** et on montre à l'utilisateur les renseignements obtenus sur lui (machine, client

2. Il est possible de placer le source du formulaire dans la procédure, elle doit alors émettre le source du formulaire si la méthode est **GET**, traiter les champs saisis si la méthode est **POST**.

WWW, ...).

```
<HTML>
<!-- Source HTML permettant d'afficher le formulaire et
      d'appeler la procédure CGI référencée par l'attribut
      action de la balise <FORM>.
-->
<HEAD>
<TITLE> Test CGI </TITLE>
</HEAD>
<BODY>
<H1> Test CGI </H1>
<HR>
<FORM action="http://madeo.irisa.fr/cgi-bin/tstcgi"
      method="POST">
  Nom :   <INPUT TYPE="text" NAME="nom"      SIZE="25"> <BR>
  Prénom: <INPUT TYPE="text" NAME="prenom"  SIZE="15"> <BR>
  <INPUT TYPE="submit" VALUE="OK">
  <INPUT TYPE="reset" VALUE="Annuler">
</FORM>
<HR>
</BODY>
</HTML>
```

16.2.1 Utiliser cgi-lib.pl

La première bibliothèque qui a été diffusée pour écrire des scripts CGI en Perl est cgi-lib.pl³. Elle reste aujourd'hui très utilisée et présente les caractéristiques suivantes :

- permet de récupérer les champs transmis dans un tableau associatif ;
- gère le transfert de fichiers dans le sens client serveur (*upload*) ;
- globalement très simple à utiliser.

Le script Perl suivant propose d'utiliser cgi-lib.pl pour réaliser notre exemple, il utilise les fonctions suivantes :

- **Methpost()** qui rend vrai si la méthode HTTP utilisée est **POST**;

3. <http://cgi-lib.stanford.edu/cgi-lib/>

- **ReadParse()** qui prend une référence sur un hash et le complète par une association entre le nom des champs (tel que spécifiés par l'attribut NAME de la balise <FORM>) et leurs valeurs respectives;
- **PrintHeader()** qui indique au navigateur qu'il va recevoir un document HTML. Cette fonction prend aussi à sa charge l'émission d'une ligne vide pour séparer les en-têtes du corps de document (cf. RFC2068 sur HTTP/1.1);
- **CgiDie** pour quitter la procédure en émettant un message d'avertissement à l'utilisateur;
- ...

```
#!c:\perl\bin\perl.exe
# TestCGI utilise cgi-lib pour acquerir les champs du
# formulaire et generer le HTML resultat. On verifie
# que la methode est POST, que les champs ne sont pas
# vides et on informe l'utilisateur du resultat.

require "cgi-lib.pl";
$champs= {}; # $champs est une reference sur un hash vide

if (MethPost()) {
    ReadParse($champs);
    if ($champs->{"nom"} eq "" || $champs->{"prenom"} eq "")
    CgiDie("Il faut remplir les champs !");
}
print (PrintHeader(),
       HtmlTop("Resultat de votre requete"),
       "Nom : ", $champs->{"nom"}, "<BR>",
       "Prenom : ", $champs->{"prenom"}, "<BR>",
       "vous utilisez $ENV{HTTP_USER_AGENT}",
       "depuis la machine $ENV{REMOTE_ADDR}",
       HtmlBot());
}
else {
    CgiDie("Hum ... Que faites vous !");
}
```

16.2.2 Utiliser CGI.pm

CGI.pm⁴ est un autre outil facilitant la réalisation de script CGI en Perl. CGI.pm peut être utilisé de deux façons :

- en important des fonctions dans l'espace du script, CGI.pm est alors utilisé comme un package;
- en utilisant un objet de classe CGI comme dans la réalisation (ci-après) de notre exemple.

```
#!/c:\perl\bin\perl.exe
# TestCGI utilise CGI.pm pour acquérir les champs du
# formulaire et generer le HTML resultat. On verifie
# que la methode est POST, que les champs ne sont pas
# vides et on informe l'utilisateur du resultat.
use CGI;
$req = new CGI;

print $req->header();
print $req->start_html('Résultat de la requête');
if ($req->request_method() eq "POST") {
    if ($req->param('nom') eq " " ||
        $req->param('prenom') eq " " ) {
        print $req->h1('Il faut remplir les champs');
    }
    else {
        print ("Nom : ", $req->param('nom'),
            $req->br,
            "Prenom : ", $req->param('prenom'),
            $req->br,
            " vous utilisez", $req->user_agent(),
            "depuis la machine ", $req->remote_addr());
    }
}
else {
print $req->h1('Hum ... Que faites vous !');
}
print $req->end_html;
```

4. <http://www.wiley.com/compbooks/stein/index.html>

CGI.pm peut sembler un peu plus lourd à utiliser que `cgi-lib.pl`, il est toutefois plus élaboré pour :

- générer les en-têtes HTTP, dans notre exemple, `$req->header` se contente de positionner `Content-Type: text/html` (suivi d'une ligne vide), mais pourrait aussi indiquer une en-tête `expire` (`$req->header(-expires=>'now');`) pour inhiber les caches;
- générer du code HTML (notamment des tables, des formulaires, ...);
- être utilisé dans un contexte FastCGI (cf. 16.4);
- mettre au point les scripts sans passer par un navigateur WWW et un serveur HTTP. CGI.pm reconnaît s'il est appelé dans un contexte dit *offline* et invite alors l'utilisateur à rentrer à la main des couples `clef=valeur` pour simuler la chaîne complète d'un environnement CGI.

16.3 Envoyer un fichier (*upload*)

Il est parfois intéressant de transférer des documents complets depuis un client WWW vers un serveur HTTP. CGI.pm ainsi que `cgi-lib.pl` offrent cette possibilité. Elle s'appuie sur le transfert dans le corps d'un document (comportant plusieurs parties), des champs éventuels d'un formulaire suivi des différents fichiers.

On peut compléter notre exemple en demandant à l'utilisateur de fournir deux fichiers qui seront transférés sur le serveur et transmis à notre script (via `cgi-lib.pl` ici). Le code HTML peut être modifié de la façon suivante :

```
<HTML>
<!-- Le document est composé de plusieurs parties, donc
      (ENCTYPE="multipart/form-data") + ajout de 2 champs "file".
-->
<HEAD> <TITLE> Test CGI </TITLE> </HEAD>
<BODY>
<H1> Test CGI </H1><HR>
<FORM action="http://madeo.irisa.fr/cgi-bin/tfic.pl"
      ENCTYPE="multipart/form-data" method="POST">
  Nom : <INPUT TYPE="text" NAME="nom" SIZE="25"> <BR>
  Prénom: <INPUT TYPE="text" NAME="prenom" SIZE="15"> <BR>
  <INPUT TYPE="file" NAME="fic1" SIZE=50> <BR>
  <INPUT TYPE="file" NAME="fic2" SIZE=50> <BR>
  <INPUT TYPE="submit" VALUE="OK">
  <INPUT TYPE="reset" VALUE="Annuler">
</FORM> </BODY> </HTML>
```

Dans le cas d'un type de document *multipart/form-data* les arguments de la fonction `ReadParse` sont 4 références à des tableaux associatifs correspondants aux champs du formulaire, aux nom des fichiers transmis, aux *Content-Type* des fichiers transmis et aux noms des fichiers reçus. Notre procédure peut donc être modifiée de la façon suivante :

```
#!c:\perl\bin\perl.exe
#
# Réception des deux fichiers.

require "cgi-lib.pl";
$champs = {};          # $champs, $client_fn,
$client_fn = {};      # $client_ct, $serveur_fn
$client_ct = {};      # sont des références sur
$serveur_fn = {};    # des hashes vides.
$cgi_lib::writefiles = "c:\tmp"; # Répertoire dans lequel
                                # les fichiers transmis
                                # seront déposés.

if (MethPost()) {
  ReadParse ($champs, $client_fn, $client_ct, $serveur_fn);
  if ($champs->{"nom"} eq "" || $champs->{"prenom"} eq "") {
    CgiDie("Il faut remplir les champs !");
  }
  print (PrintHeader(),
         HtmlTop("Resultat de votre requete"),
         "Nom : ", $champs->{"nom"}, "<BR>",
         "Prenom :", $champs->{"prenom"}, "<BR>",
         $serveur_fn->{'fic1'}, # nom du fichier 1 (serveur)
         $serveur_fn->{'fic2'}, # nom du fichier 2 (serveur)
         $client_fn->{'fic1'},  # nom du fichier 1 (client)
         $client_fn->{'fic2'},  # nom du fichier 2 (client)
         $client_ct->{'fic1'},  # Content-Type fichier 1
         $client_ct->{'fic2'},  # Content-Type fichier 2
         HtmlBot());
  # Une procedure opérationnelle pourrait maintenant ouvrir
  # les fichiers correspondants et effectuer un traitement.
}
else {
  CgiDie("Hum ... Que faites vous !");
}
```

Pour qu'une procédure effectuant du transfert de fichiers se déroule normalement, il convient que le serveur HTTP soit autorisé à écrire dans le répertoire `$cgi_lib::writefiles`,

il faut également que celui-ci soit correctement dimensionné.

16.4 Utiliser FastCGI

16.4.1 Introduction

Pour contourner le manque de performance des scripts CGI écrits en Perl, deux solutions sont envisageables :

- utiliser `mod_perl`⁵ qui lie APACHE⁶ et Perl et permet d'éviter la création de processus à chaque appel de procédure. Cette solution présente l'avantage de permettre l'écriture en Perl de modules pour APACHE⁷, mais elle introduit quelques inconvénients, notamment une très forte augmentation de la taille mémoire utilisée par un processus *httpd*;
- utiliser FastCGI⁸ qui fait communiquer le serveur HTTP et la procédure externe via une *socket*. La procédure externe fonctionne alors comme un serveur à l'écoute sur un port, elle traite les requêtes de son client (le serveur HTTP) qui lui transmet les données émises par les navigateurs WWW. La procédure externe est démarrée à l'initialisation du serveur HTTP, elle est persistante, le gain de performance par rapport à CGI est très significatif.
FastCGI permet également de délocaliser la procédure externe (la faire se dérouler sur une machine distincte de celle qui héberge le serveur HTTP), c'est une fonctionnalité très intéressante, elle permet de répartir les traitements.

CGI.pm permet d'utiliser FastCGI comme canal de communication, en conséquence les changements à apporter pour faire migrer une procédure de CGI vers FastCGI sont très minimes (cf. exemple ci-dessous).

16.4.2 Un exemple

On reprend ici l'exemple de la section 16.2.2, les changements à effectuer ont pour but d'indiquer à CGI.pm de basculer dans un contexte FastCGI et de rendre la procédure

5. Consulter <ftp://ftp.bora.net/pub/CPAN/modules/by-module/Apache/> la version actuelle est `mod_perl-1.18.tar.gz`

6. Le serveur HTTP le plus utilisé au monde, consulter <http://www.apache.org>

7. Pour ajouter des fonctionnalités au serveur HTTP de base.

8. Consulter <http://www.fastcgi.com/> et installer le module `fastcgi` pour APACHE ainsi que FCGI pour Perl

persistante (boucler sur l'attente de l'arrivée de données à traiter). Pour mettre en évidence l'effet de la persistance de la procédure, on ajoute un compteur de requêtes.

```
#!/usr/local/bin/perl
# TestCGI utilise CGI.pm pour acquerir les champs du
# formulaire et generer le HTML resultat. On verifie
# que la methode est POST, que les champs ne sont pas
# vides et on informe l'utilisateur du resultat.

use CGI::Fast;
$Cptr = 0;
while ($req=new CGI::Fast) {
    $Cptr++;
    print $req->header();
    print $req->start_html('Résultat de la requête');
    if ($req->request_method() eq "POST") {
        if ($req->param('nom') eq " " ||
            $req->param('prenom') eq " ") {
            print $req->h1('Il faut remplir les champs');
        }
        else {
            print ("Requête numéro : ", $Cptr, $req->br);
            print ("Nom : ", $req->param('nom'),
                $req->br,
                "Prenom : ", $req->param('prenom'),
                $req->br,
                " vous utilisez", $req->user_agent(),
                "depuis la machine ",$req->remote_addr());
        }
    }
    else {
        print $req->h1('Hum ... Que faites vous !');
    }
    print $req->end_html;
}
```


Chapitre 17

La programmation d'applications réseaux

17.1 Introduction

Perl permet le développement d'applications utilisant les *sockets*. Il s'agit de canaux de communications utilisant un couple *numéro IP/numéro de port* pour mettre en rapport deux machines. Le numéro IP cible une machine sur le réseau, le numéro de port adresse une application sur la machine référencée. On peut faire communiquer des processus distants en utilisant des *sockets* TCP ou UDP. TCP assure un service de communication fiable entre applications (remise en ordre des paquets, demande de réémission des paquets corrompus), UDP est beaucoup plus léger (pas de contrôles sur la validité des paquets) et est utilisé dans des cas bien spécifiques (transmission audios et vidéos où les contrôles sur les paquets n'ont pas de sens, ...).

On peut également utiliser les *sockets* pour faire communiquer des processus résidant sur une même machine (*sockets* dites Unix), pour plus de détails, consulter le manuel des fonctions socket, bind, listen, accept ...

17.2 Un exemple d'application client/serveur

L'exemple suivant permet de faire communiquer un serveur et des clients en utilisant une *socket* TCP. Le serveur est à l'écoute sur le port 8888, il reçoit et stocke des messages en provenance de clients distants.

Le code qui suit n'est pas à sortir du contexte de cet exemple, une véritable application devrait utiliser un service TCP enregistré, acquitter les messages reçus, ...

```
#!/usr/local/bin/perl
# Serveur qui réceptionne sur le port 8888 des messages
# qu'il stocke dans un fichier.

use Socket;
my $port = 8888;
my $protoc = getprotobyname('tcp');

open (FICLOG,">Log") || die ("Erreur Création Log : $!");
# Création de la socket TCP de reception des messages
socket (SockPerm, PF_INET, SOCK_STREAM, $protoc)
    || die("Erreur Creation Socket : $!");
setsockopt (SockPerm, SOL_SOCKET, SO_REUSEADDR, 1)
    || die ("setsockopt : $!");
bind (SockPerm, sockaddr_in($port, INADDR_ANY))
    || die("Erreur bind : $!");
listen (SockPerm, SOMAXCONN);

$SIG{INT}='arret'; # Fermer le fichier des messages si ^C
# Réception des messages
while(1) {
    $refcli = accept(SockTmp,SockPerm);
    my ($port, $addr) = sockaddr_in($refcli);
    $client=gethostbyaddr($addr, AF_INET);
    #
    # Traitement d'un message
    #
    $ligne=<SockTmp>;
    chop($ligne);
    until ($ligne =~ /^FIN/) {
        print FICLOG ("$client : ",$ligne,"\n");
        $ligne=<SockTmp>;
        chop($ligne);
    }
    close(SockTmp);
}
sub arret {
    close(FICLOG); exit;
}
```

```
#!/usr/local/bin/perl
#
# Client qui emet les messages.
#
use Socket;

my $port = 8888;
my $protoc = getprotobyname('tcp');
my $adrserv = 'localhost'; # machine locale pour les tests
my $iaddr = inet_aton($adrserv);
my $paddr = sockaddr_in($port,$iaddr);
#
# création de la socket et connexion au serveur
#
socket (SOCK, PF_INET, SOCK_STREAM, $protoc)
    || die ("Erreur Creation Socket: $!");
connect(SOCK, $paddr) || die ("Erreur Connect: $!");
#
# Saisie et émission ligne à ligne du message
#
while (<STDIN>) {
    print SOCK ("$_");
}
#
# Emission de la marque de fin de message
#
print SOCK ("FIN \n");
close(SOCK);
```


Bibliographie

- [1] Larry WALL Tom CHRISTIANSEN et Randal L. SCHWARTZ.
Programmation en Perl, 2ème édition en français.
O'REILLY, 1996. ISBN 2-84177-004-4.
- [2] Sriram SRINIVASAN.
Advanced Perl Programming.
O'REILLY, 1997. ISBN 1-56592-220-4.
- [3] Randal L. SCHWARTZ Erik OLSON and Tom CHRISTIANSEN.
Learning Perl on Win32 Systems.
O'REILLY, 1997. ISBN 1-56592-324-3.
- [4] Bruno POULIQUEN.
Introduction au langage Perl.
<http://www.med.univ-rennes1.fr/~poulique/cours/perl/>.
- [5] Olivier AUBERT.
Introduction à perl.
<http://perso-info.enst-bretagne.fr/~aubert/perl/html/perl.html>.